
Section 1 Examples from Three Data Sets

Data Set 1: The COMBINE Study of Alcohol Use Disorder

Data from COMBINE, a 16-week, multisite randomized
double-blind clinical trial comparing treatments of alcohol
dependence (Anton et al., 2006, JAMA)

N = 1,383. Mean age 44
Measurement occasions: Baseline, week 1, week 2, week 4, week
6, week 8, week 10, week 12, week 16 and week 52 follow-up

Stress: Brief version of The Perceived Stress Scale

Alcohol risk: Abstinence, low risk, medium risk, high risk, very
high risk

Heavy Drinking: Number of heavy drinking days per week
Covariates:

Intervention - 9 groups (medication, placebo, and therapy),
gender, race, age, education, marital status, employment

Stress and alcohol use disorder (AUD). Stress causes drinking
(Armeli et al., 2000 in J of Personality and Social Psych)

Distribution of the Stress Variable (Week 1)

A 4-item version of The Perceived Stress Scale with scores of 0 to 16
has been used for analyses of the COMBINE data:

McHugh et al. (2013). Positive affect and stress reactivity in
alcohol-dependent outpatients. J. Studies in Alcohol and Drugs

Can be treated as a continuous variable using MLR

Percentage at the lowest score of zero increases with time but does not
exceed 15 %

Distribution of the Alcohol Risk Variable

50%

25%

9% 9%
7%

Abstinence Low Risk Medium Risk High Risk Very High Risk

WHO categories based on grams of pure alcohol per day (separate for
males and females)

Should not be treated as a continuous variable with linear relations due
to strong floor effect: Biases in correlations and regressions

Ordered categorical (ordinal) with 5 categories. Floor effect not a
problem

Binary: Abstinence or not

Distribution of Heavy Drinking Days Per Week

66%

11%
6% 6%

76543210

4% 2% 2%3%

Number of Heavy Drinking Days Per Week

Not really a count variable, but can be treated as:

Ordered categorical (ordinal) variable
Dichotomized variable
Censored variable
Two-part variable

Data Set 2: Suicidal Ideation and Substance Abuse

Suicidal ideation and substance abuse

Classic question of what influences what

Data from a preventive intervention study in Baltimore (Ialongo)

N = 737

T = 8: Ages 19-26

Covariates: Gender, race, lunch (poverty indicator)
References:

Ialongo, Werthamer, Kellam, Brown, & Wang (1999). Proximal
impact of two first grade preventive interventions on the early risk
behaviors for later substance abuse, depression, and antisocial
behavior. American Journal of Community Psychology, 27,
599-641
Musci et al. (2016). Suicide & Life Threatening Behavior
Thrul et al. (2021). Addiction

Suicidal Ideation and Substance Abuse: Binary Outcomes

At age 19, 77% are at zero, 8% at 1: Dichotomize into 0 vs higher

At least one suicidal ideation and/or behaviors endorsed in the
last year (Y)
At least one substance abuse or dependence criteria met across all
substances assessed in the last year (Z)

Age 19 Age 20 Age 21 Age 22 Age 23 Age 24 Age 25 Age 26

Y 23.0% 18.2% 15.2% 19.0% 23.0% 22.3% 23.4% 21.1%

Z 19.9% 19.8% 15.3% 20.4% 18.5% 15.0% 14.3% 13.3%

Data Set 3: Negative Affect

Data from the older cohort of the Notre Dame Study of Health &
Well-being (Bergeman): N = 271, T = 56 (daily measures on
consecutive days)

10 NA items (5-category scale): afraid, ashamed, guilty, hostile,
scared, upset, irritable, jittery, nervous, distressed

Question format: Today I felt... (1 = Not at all, 2 = A little, 3 =
Moderately, 4 = Quite a bit, 5 = Extremely)

Wang, Hamaker, Bergeman (2012). Investigating inter-individual
differences in short-term intra-individual variability. Psychological
Methods

Negative Affect: Ordered Categorical (Ordinal) Items

Question format: Today I felt... (1 = Not at all, 2 = A little, 3 =
Moderately, 4 = Quite a bit, 5 = Extremely)

Irritable item: 68% at lowest value - Not at all

Not suitable for continuous variable analysis due to strong floor effect

Can be treated as an ordinal variable

Section 2 Refresher of Mplus Web Talk 4 Part 1
Analysis of Continuous Outcomes

Dynamic Models

Auto-Regression of lag 1 (AR1)

y1 y2 y3 y4 y5 y6 y7 y8

c3c1 c2 c4 c5 c6 c7 c8

i

y9 y10

Dynamic Random Intercept AR1 (D-RI-AR1). Bollen-Brandt (2010)

y1 y2 y3 y4 y5 y6 y7 y8 y9 y10

i

Dynamic Random Intercept ARMA (1,1) (D-RI-ARMA11). Zyphur et
al. (2020)

y1 y2 y3 y4 y5 y6 y7 y8 y9 y10

i

Random Intercept and Auto-Correlated Residuals
RI-AR Modeling Displayed in Two Equivalent Ways

y1 y2 y3 y4 y5 y6 y7 y8

c3c1 c2 c4 c5 c6 c7 c8

i

y9 y10

0 0 0 0 0

0 0 0 0 0 0 0 0 0 0

i

y1 y2 y3 y4 y5 y6 y7 y8 y9 y10

y1 y2 y3 y4 y5 y6 y7 y8 y9 y10

i

y1^ y2^ y3^ y4^ y5^ y6^ y7^ y8^ y9^ y10^

y1 y2 y3 y4 y5 y6 y7 y8 y9 y10

RI-AR modeling is the univariate part of RI-CLPM

Time-State-Error (TSE) model allows restricted measurement error

CLPM does not include the random intercept i, so using AR1

RI-ARMA Modeling (Asparouhov-Muthén, 2022; RSEM)

y1 y2 y3

f1 f2 f3

i

0 0 0

y4 y5

f4 f5

0 0

y6 y7

f6 f7

0 0

y8 y9

f8 f9

0 0

y10

f10

0

This model is similar in spirit to RI-AR because of its separation of
between- and within-individual variation also referred to as latent
centering (centering using the random intercept i), but adds an MA
component

An equivalent measurement error version, RI-MEAR, is available
which is more general than TSE but like TSE often presents estimation
problems not seen with RI-ARMA

RI and AR1 Impact on Correlations Across Time (T = 10)

(a) RI variance (R2) = 0.00 (b) RI variance (R2) = 0.25

(c) RI variance (R2) = 0.50 (d) RI variance (R2) = 0.75

Correlation Y1,Yt = ψ +β t−1(1−ψ) where ψ is the random intercept variance, β is the
constant auto-regression among the residuals, and Y variances are all 1

COMBINE Stress and Alcohol Risk: Baseline - Week 16

Stress
Means

Week
0

4

5

6

1

2

3

10 12 14 160 2 4 6 8

0

0.4

0.5

0.6

Week

0.1

0.2

0.3

10 12 14 160 2 4 6 8

Bengt Muthén RI-CLPM 18/ 162



COMBINE Data Stress Outcome:
Covariance Coverage for Week 1 - Week 16 (N=1375)

Week 1 = y1, week 2 = y2, week 4 = y3, week 6 = y4, week 8 = y5,
week 10 = y6, week 12 = y7, week 16 = y8

Y1 Y2 Y3 Y4 Y5 Y6 Y7 Y8

Y1 0.924
Y2 0.860 0.915
Y3 0.823 0.828 0.870
Y4 0.761 0.767 0.760 0.800
Y5 0.764 0.771 0.758 0.736 0.811
Y6 0.673 0.683 0.671 0.666 0.677 0.707
Y7 0.710 0.719 0.708 0.693 0.704 0.660 0.752
Y8 0.758 0.763 0.749 0.716 0.742 0.665 0.713 0.817

0.817 says that 81.7% have data on Y8 (don’t have missing on Y8)

0.713 says that 71.3% have data on both Y8 and Y7 (don’t have
missing on either Y8 or Y7)

Average Correlations for the Stress Outcome
as a Function of the Time Distance

0

0.1

0.2

0.3

0.4

0.5

0.6

0.7

0.8

0.9

1

1 3 5 7 9 11 13 15

Average 
Correlation

Time Distance in Weeks

Stress

Weeks observed: 1, 2, 4, 6, 8, 10, 12, 16 (T=8)

Number of correlations at each time distance in weeks:
1 at 1, 5 at 2, 1 at 3, 5 at 4, 1 at 5, 4 at 6, 1 at 7, 3 at 8, 1 at 9, 2 at
10, 1 at 11, 1 at 12, 0 at 13, 1 at 14, 1 at 15 (28 correlations)

Plot suggests high RI variance and sizeable AR

Univariate Analysis of Stress using MLR (N=1375, T=8)

Outputs for all models are posted on the website of the talk

Model # par’s LL BIC Chi-square RMSEA CFI

1. AR1 23 -20409 40984 χ2(21)=747 0.160 0.818
(.0000) (<.05=.000)

2. AR2 29 -20062 40334 χ2(15)=233 0.104 0.945
(.0000) (<.05=.000)

3. ARMA11 29 -19921 40051 χ2(15)=18 0.013 0.999
(.2432) (<.05=1.000)

4. D-RI-AR1 24 -19964 40102 χ2(20)=86 0.049 0.983
(.0000) (<.05=.520)

5. D-RI-AR2 30 -19946 40108 χ2(14)=59 0.048 0.989
(.0000) (<.05=.553)

6. D-RI-ARMA11* 31 -19918 40060 χ2(13)=14 0.007 1.000
(.3787) (<.05=1.000)

7. RI-AR1 24 -19951 40076 χ2(20)=72 0.044 0.987
(.0000) (<.05=.819)

8. RI-AR2 30 -19928 40073 χ2(14)=33 0.032 0.995
(.0027) (<.05=.983)

9. RI-ARMA 30 -19919 40054 χ2(14)=15 0.008 1.000
(.3616) (<.05=1.000)

* Negative insignificant V(i)

Mplus Input for RI-ARMA

y1 y2 y3

f1 f2 f3

i

0 0 0

y4 y5

f4 f5

0 0

y6 y7

f6 f7

0 0

y8 y9

f8 f9

0 0

y10

f10

0

Model:
i BY y1-y8@1;

f1 BY y1;
f2 BY y2;
f3 BY y3;
f4 BY y4;
f5 BY y5;
f6 BY y6;
f7 BY y7;
f8 BY y8;

i WITH f1@0; ! or, MODEL=NOCOV

y1-y8@0;

f2-f8 PON f1-f7; ! AR part
f3-f8 PON f2ˆ-f7ˆ; ! MA part

f’s could be replaced by yˆ, but then we would need yˆˆ for the moving average
part f3-f8 PON f2ˆ-f7ˆ;

Model 9 RI-ARMA Estimates Using MLR

Substantial random
intercept variance =
3.924, S.E. = 0.900

STDYX estimates:

Estimate S.E. Est./S.E.

I BY
Y1 0.672 0.073 9.157
Y2 0.636 0.075 8.466
Y3 0.628 0.075 8.336
Y4 0.631 0.072 8.743
Y5 0.626 0.073 8.613
Y6 0.636 0.070 9.076
Y7 0.619 0.068 9.047
Y8 0.637 0.075 8.527

Corr (Y1, Y8) due to RI =
0.672 × 0.637 = 0.428
(total est corr = 0.539)

R2 (Y1) = 0.6722 = 0.452

STDYX estimates:

Estimate S.E. Est./S.E.

F2 ON
F1 0.522 0.086 6.095

F3 ON
F2 0.836 0.082 10.192
F2ˆ -0.345 0.063 -5.485

F4 ON
F3 0.759 0.096 7.942
F3ˆ -0.324 0.072 -4.505

F5 ON
F4 0.809 0.102 7.964
F4ˆ -0.441 0.074 -5.971

F6 ON
F5 0.820 0.088 9.265
F5ˆ -0.432 0.071 -6.099

F7 ON
F6 1.039 0.074 14.006
F6ˆ -0.622 0.095 -6.544

F8 ON
F7 0.848 0.105 8.040
F7ˆ -0.483 0.081 -5.973

Section 3 Brief Introduction
to Analysis of Binary Outcomes

Brief Introduction to Modeling with Binary Outcomes (0/1)

P (yj = 1 | f )
1

0 f

y1 y2 y3 y4 y5

f

y1 y2 y3 y4 y5

f

P (yj = 1 | f )
1

0 f

Three contexts:
Item Response Theory (IRT): item difficulty and discrimination
Factor analysis: item thresholds and factor loadings
Random intercept: factor loadings fixed at 1

Typical specification:
Normally distributed latent variable
Logistic or Probit regressions

See Short Course Topic 2 and WLSMV, ML, and Bayes estimation at
https://www.statmodel.com/download/
EstimatorChoices.pdf

Brief Introduction to Modeling with Binary Outcomes

Equivalent representation with continuous latent response variables Y*:

y1 y2 y3 y4 y5

y1* y2* y3* y4* y5*

f

Y* > threshold results in Y = 1, otherwise Y = 0

Specifying normally distributed f together with probit regressions is the
same as specifying normally distributed latent response variables Y*
(N + N = N). Logistic regression does not give Y* normality

Correlations between Y* variables for binary Y: Tetrachoric
correlations

Brief Introduction Continued: Different Interests

The cross-sectional modeling of IRT and factor analysis focuses on the
relationship between f and Y

The longitudinal modeling of panel data analysis adds a focus on the
relationship between Y’s at different time points

The factor cannot account for all the correlation among the Y*’s
Multivariate probit modeling allows linear regressions among
normally distributed Y*’s such as in this D-RI-AR1 model:

y1 y2 y3 y4 y5

i

y1* y2* y3* y4* y5*

Bengt Muthén RI-CLPM 27/ 162



Brief Introduction Continued: Ordinal Outcomes
Polychoric Correlations

59

St
ro

ng
ly

Ag
re

e

y j*
yj

St
ro

ng
ly

D
is

ag
re

e

Strongly
Disagree

Strongly
Agree

y k*

117

yk

The ordinal case still allows linear regression between Y*’s (as
opposed to the nominal case)

General latent response variable modeling assuming underlying
normality for Y* available in Mplus for WLSMV and Bayes

Two Kinds of Model Assessments
for Categorical Outcomes

Fit to correlations among a set of normal, continuous latent
response variables Y∗ underlying the observed categorical Y’s
(WLSMV and Bayes)

Muthén et al. (1997). Robust inference using weighted least
squares and quadratic estimating equations in latent variable
modeling with categorical and continuous outcomes.
Unpublished technical report. http://www.statmodel.
com/download/Article_075.pdf

Fit to the data in the form of response patterns, that is, a
frequency table for all variables

A model may fit the Y∗ correlations but not the frequency table
Even a just-identified Y∗s model with all WITH’s may not fit the
frequency table in some cases
Muthén (1993). Goodness of fit with categorical and other
non-normal variables. In K.A. Bollen, & J.S. Long (Eds), Testing
Structural Equation Models (pp. 205-243). Newbury Park, CA:
Sage http://www.statmodel.com/bmuthen/
articles/Article_045.pdf

Fit to Y* Correlations for Categorical Variables

Fit to correlations among a set of normal, continuous latent
response variables Y∗ underlying the observed categorical Y’s
(WLSMV and Bayes)

The Muthén et al. (1997) WLSMV chi-square works well when
the number of variables is not large and the sample size is not
small: Suitable for cross-lagged panel modeling
Bayes PPP idea: Using any fit statistic, compute the fit statistic
for the observed data, generate a fit statistic distribution based on
generated data from the estimated model, and find the proportion
of cases where the latter is larger than the former

Bayes PPP for categorical variables: Based on chi-square test of
overall model fit for Y∗s
Analogous to WLSMV chi-square test of estimated versus sample
tetrachorics and polychorics
Low power for binary outcomes and less powerful than the
WLSMV chi-square test (Asparouhov-Muthén, 2021a)
More powerful for polytomous variables

Fit to the Data in Terms of Frequency Tables

Frequency table test of model fit:
With categorical latent class indicators, the model can be tested
against data using Pearson and likelihood-ratio chi-square
frequency table tests. Summing over the cells of the table:

Pearson : ∑
j
(oj − ej)

2/ej

Likelihood ratio : 2∑
j

oj log(oj/ej)

There are typically too many frequency table cells with many
cells having estimated frequencies close to zero, invalidating the
tests: Pearson and Likelihood ratio tests disagree
Example with 8 binary variables: 28 = 256 possible response
patterns, where many patterns are probably not observed (zero
cells in the frequency table) - the two tests disagree strongly

Alternative checks: Fit for univariate and bivariate tables (higher freq’s)

New TECH10 for WLSMV and Bayes in 8.7 and 8.8: Standardized
residuals for response patterns, uni- and bi-variate frequency tables,
and Bayes PPP for Pearson fit to uni- and bi-variate tables

Example of Frequency Table Fit for Data Set 2
Suicidal Ideation and Substance Abuse (N = 737, T = 8)

All WITH model to estimate sample correlations:
y1-z8 WITH y1-z8;
For these 16 variables, the most frequent pattern is 0’s for both
variables at all 8 time points, which is observed for 104
individuals (14%)
WLSMV and Bayes
For an introduction to using Bayes estimation in Mplus, see
Mplus Short Course Topic 12, Parts 3-4, slides 4-21

ANALYSIS:
ESTIMATOR = WLSMV;
PARAMETERIZATION = THETA;
PROCESSORS = 8;

MODEL:
y1-y8 WITH z1-z8;

OUTPUT: RESIDUAL TECH10;
PLOT: TYPE = PLOT3;

ANALYSIS:
ESTIMATOR = BAYES;
BITERATIONS = (1000);
PROCESSORS = 8;

MODEL:
y1-y8 WITH z1-z8;

OUTPUT: TECH8 RESIDUAL TECH10;
PLOT: TYPE = PLOT3;

All WITH Model Fit for WLSMV and Bayes:
Chi-Square, PPP, Response Patterns, and Frequency Tables

MODEL FIT INFORMATION
Number of Free Parameters: 136
Chi-Square Test of Model Fit
Value 0.000*
Degrees of Freedom 0
P-Value 0.0000

MOST FREQUENT RESPONSE PATTERNS
Response Frequency Standardized
Pattern Observed Estimated Residual

(z-score)
1 104.00 98.27 0.65
2 12.00 12.69 -0.20
3 10.00 8.10 0.60
4 7.00 8.48 -0.64
5 7.00 5.44 0.59

Number of Significant Standardized Residuals 0
BIVARIATE PROPORTIONS FOR CATEGORICAL VARI-
ABLES
Overall Number of Significant Standardized Residuals 0

MODEL FIT INFORMATION
Number of Free Parameters 136
Bayesian Posterior Predictive Checking using Chi-Square
95% Confidence Interval for the Difference Between
the Observed and the Replicated Chi-Square Values

-49.528 53.388
Posterior Predictive P-Value 0.598

MOST FREQUENT RESPONSE PATTERNS
Response Frequency Standardized
Pattern Observed Estimated Residual

(z-score)
1 104.00 97.16 0.78
2 12.00 10.62 0.40
3 10.00 8.47 0.48
4 7.00 8.30 -0.54
5 7.00 3.48 1.33

Number of Significant Standardized Residuals 0
BIVARIATE PROPORTIONS FOR CATEGORICAL VARI-
ABLES
Overall Number of Significant Standardized Residuals 0

Section 4 Data Set 2 Binary Outcomes:
Suicidal Ideation and Substance Abuse

Coverage for Suicidal Ideation and/or Behaviors (Y)
Dichotomized Outcomes, N = 737, T = 8

Y1 Y2 Y3 Y4 Y5 Y6 Y7 Y8

Y1 0.808
Y2 0.727 0.803
Y3 0.724 0.746 0.823
Y4 0.705 0.704 0.738 0.832
Y5 0.700 0.693 0.726 0.754 0.832
Y6 0.711 0.707 0.735 0.751 0.776 0.844
Y7 0.701 0.701 0.727 0.739 0.754 0.774 0.848
Y8 0.700 0.702 0.730 0.730 0.731 0.754 0.780 0.837

0.808 says that 80.8% have data on Y1 (don’t have missing on Y1)

0.727 says that 72.7% have data on both Y1 and Y2 (don’t have
missing on either Y1 or Y2)

Similar coverage for Substance abuse

The 20 Most Frequent Response Patterns (* = Missing)
Suicidal Ideation Substance Abuse

Pattern Frequency Percentage Pattern Frequency Percentage

0 0 0 0 0 0 0 0 155.00 21.0 0 0 0 0 0 0 0 0 201.00 27.3
1 0 0 0 0 0 0 0 25.00 3.4 1 1 0 0 0 0 0 0 18.00 2.4
0 1 0 0 0 0 0 0 14.00 1.9 0 0 0 1 0 0 0 0 15.00 2.0
* 0 0 0 0 0 0 0 12.00 1.6 0 0 1 0 0 0 0 0 14.00 1.9
0 0 0 0 0 0 0 * 11.00 1.5 * 0 0 0 0 0 0 0 12.00 1.6
0 0 0 0 0 0 0 1 11.00 1.5 0 0 0 * 0 0 0 0 10.00 1.4
0 0 0 0 0 0 1 1 10.00 1.4 0 0 0 0 0 0 0 * 9.00 1.2
0 * 0 0 0 0 0 0 9.00 1.2 0 0 0 0 1 0 0 0 8.00 1.1
0 0 0 1 0 0 0 0 9.00 1.2 0 * 0 0 0 0 0 0 7.00 1.0
0 0 0 0 0 0 1 0 8.00 1.1 0 0 0 0 0 0 * 0 7.00 1.0
* * 0 0 0 0 0 0 7.00 1.0 0 0 0 0 0 0 0 1 7.00 1.0
0 0 0 0 0 0 * 0 6.00 0.8 0 0 0 0 0 0 1 0 7.00 1.0
0 0 0 0 * 0 0 0 6.00 0.8 * * 0 0 0 0 0 0 7.00 1.0
1 0 0 0 1 0 0 0 6.00 0.8 1 1 1 1 1 1 1 1 6.00 0.8
0 0 0 0 1 0 0 0 6.00 0.8 1 0 0 0 0 0 0 0 6.00 0.8
0 0 0 * * * * * 5.00 0.7 * * * * * * 0 0 6.00 0.8
0 0 0 * 0 0 0 0 5.00 0.7 0 0 0 0 * 0 0 0 6.00 0.8
0 * * * * * * * 5.00 0.7 0 0 0 0 0 0 * * 6.00 0.8
0 0 0 0 0 0 * * 5.00 0.7 0 * * * * * * * 6.00 0.8
0 0 1 0 0 0 0 0 4.00 0.5 * * * 0 0 0 0 0 6.00 0.8

Percentage 0’s or missing: 43.2% and 53.5%, respectively.

Sample Correlations for Suicidal Behavior
and Substance Abuse (Bayes; N = 737, T = 8)

Tetrachoric correlations using an unrestricted, all WITH model:
y1-z8 WITH y1-z8;

Standard errors ranging from 0.05 to 0.10

Correlations for Suicidal Ideation (Y)

Y1 Y2 Y3 Y4 Y5 Y6 Y7 Y8
Y1 1.000
Y2 0.269 1.000
Y3 0.422 0.420 1.000
Y4 0.407 0.348 0.604 1.000
Y5 0.553 0.357 0.542 0.635 1.000
Y6 0.400 0.242 0.536 0.560 0.646 1.000
Y7 0.337 0.197 0.450 0.500 0.649 0.604 1.000
Y8 0.233 0.267 0.495 0.567 0.540 0.506 0.721 1.000

Sample Correlations Continued

Correlations for Substance Abuse (Z)

Z1 Z2 Z3 Z4 Z5 Z6 Z7 Z8
Z1 1.000
Z2 0.931 1.000
Z3 0.501 0.471 1.000
Z4 0.491 0.454 0.546 1.000
Z5 0.507 0.430 0.508 0.652 1.000
Z6 0.513 0.435 0.476 0.705 0.664 1.000
Z7 0.424 0.388 0.512 0.659 0.610 0.732 1.000
Z8 0.430 0.375 0.420 0.466 0.564 0.616 0.646 1.000

Bengt Muthén RI-CLPM 38/ 162



Sample Correlations Continued

Correlations for Y and Z: Y → Z

Y1 Y2 Y3 Y4 Y5 Y6 Y7 Y8
Z1 0.357 0.122 0.123 0.234 0.293 0.253 0.271 0.179
Z2 0.325 0.139 0.072 0.178 0.226 0.235 0.235 0.153
Z3 0.137 0.023 0.242 0.144 0.278 0.282 0.314 0.178
Z4 0.182 0.266 0.355 0.346 0.325 0.432 0.368 0.372
Z5 0.068 0.182 0.097 0.309 0.370 0.370 0.377 0.344
Z6 0.176 0.278 0.334 0.272 0.304 0.490 0.359 0.272
Z7 0.133 0.125 0.279 0.238 0.203 0.494 0.509 0.398
Z8 -0.031 0.057 0.306 0.251 0.177 0.346 0.400 0.403

Sample Correlations Continued

Correlations for Y and Z: Z → Y

CLPM Estimates

Bayes CLPM (AR11)

Y1 Y2 Y3 Y4 Y5 Y6 Y7 Y8
Z1 0.01
Z2 -0.02 0.17*
Z3 -0.01 0.03
Z4 0.26* 0.17*
Z5 0.06 0.21*
Z6 0.04 0.05
Z7 0.12 0.04
Z8 0.05

RI-CLPM cross-lagged effects are not expected to be significant unless
CLPM estimates are, the exception being cases with large ICC (e.g. 2/3
or 0.75 as in Hamaker et al., 2015)

ICC = random intercept variance/(total variance)
Total variance = random intercept variance + residual variance
With time-invariant parameters, ICC = R-square (Y) due to the
random intercept = Corr (Y1, YT) for large T

Average Tetrachoric Correlations Across Time

The plot suggest a lower asymptote larger than zero: Random
intercept variance (but much smaller than for the stress outcome
in the COMBINE data)

The plot suggests a higher AR and a higher random intercept
variance for Substance abuse than for Suicidal ideation

Section 5 Transitioning from Continuous to
Categorical Outcomes

and from Old to New Mplus Language for
RI-CLPM

Random Intercept and Auto-Correlated Residuals
RI-AR1 Modeling Displayed in Two Equivalent Ways

Continuous Outcomes

Old factor specification:

y1 y2 y3 y4

i

y5

0 0 0 0 0

Lorem ipsum
New residual (hats) specification:

y1 y2 y3 y4

i

y5

Random Intercept and Auto-Correlated Residuals
RI-AR1 Modeling Displayed in Two Equivalent Ways

Categorical Outcomes

y1 y2 y3 y4

i

y5

0 0 0 0 0

y1* y2* y3* y4* y5*

i

y1 y2 y3 y4 y5y1* y2* y3* y4* y5*

WLSMV Input for Suicide RI-AR1 Model: Naive Version

CATEGORICAL = y1-y8;
ANALYSIS:

ESTIMATOR = WLSMV;
PARAMETERIZATION = THETA;
MODEL = NOCOV;

MODEL:
i BY y1-y8@1;

wy1 BY y1;
wy2 BY y2;
wy3 BY y3;
wy4 BY y4;
wy5 BY y5;
wy6 BY y6;
wy7 BY y7;
wy8 BY y8;

y1-y8@0; ! calls for Theta param’n

wy2-wy8 PON wy1-wy7;

OUTPUT: STANDARDIZED
RESIDUAL
TECH1;
MODINDICES(ALL 0);

PLOT: TYPE = PLOT3;

Naive WLSMV Problem

THE STANDARD ERRORS OF THE MODEL PARAMETER
ESTIMATES COULD NOT BE COMPUTED. THE MODEL MAY
NOT BE IDENTIFIED. CHECK YOUR MODEL. PROBLEM
INVOLVING THE FOLLOWING PARAMETER: Parameter 24,
WY8
THE CONDITION NUMBER IS -0.157D-16.

Not all within-factor variances are identified with categorical
outcomes

Basic Identification Problem with Categorical Outcomes

For a binary outcome at a given time point, there is only 1 piece
of sample information available to estimate P(Y=1) = π , namely
the proportion - the variance is not separately identified

The mean of a binary outcome is P(Y = 1) = π and the variance
is π (1−π), that is, the variance is not a separate parameter to be
estimated but is a function of the mean
The problem is that the WY1 - WY8 factor variances were
specified to be free as the default but cannot all be identified with
categorical outcomes:

Indeterminacy: For each variable, the variance can be multiplied
by a constant and the corresponding threshold divided by the
square root of that constant - the model fit is the same
A solution is to add the line WY1-WY8@1;

WLSMV and Bayes RI-AR1 Using New Hats Input Style

i

y1 y2 y3 y4 y5y1* y2* y3* y4* y5*

y1*^ y2*^ y3*^ y4*^ y5*^

ANALYSIS:
ESTIMATOR = WLSMV;
PARAMETERIZATION = THETA;

MODEL:
iy BY y1-y8@1;
y2ˆ-y8ˆ PON y1ˆ-y7ˆ;

OUTPUT: STDYX RESIDUAL TECH10;
PLOT: TYPE = PLOT3;

ANALYSIS:
ESTIMATOR = BAYES;
BITERATIONS = (2000);
THIN = 10;
PROCESSORS = 8;

MODEL:
iy BY y1-y8@1;
y2ˆ-y8ˆ PON y1ˆ-y7ˆ;

OUTPUT: TECH8 STDYX
RESIDUAL
TECH10;

PLOT: TYPE = PLOT3;

Thresholds are free and yˆ residual variances are fixed at 1 as the
default as usual in the categorical case

Section 6 Technical Aspects of Modeling with
Binary Outcomes: Model Specification,

Identification, Estimation, and Simulations

RI-AR1 Probit Model with Hats Specification: Binary Case

i

y1 y2 y3 y4 y5y1* y2* y3* y4* y5*

y1*^ y2*^ y3*^ y4*^ y5*^
ζ2 ζ3 ζ4 ζ5

y∗t continuous latent response variable at time t with threshold τt,
y∗t > τt → yt = 1, otherwise yt = 0

y∗it = αi + ŷ∗it, (1)

ŷ∗it = βt ŷ∗it−1 +ζit; t = 2, . . . ,T (2)

ŷ∗i1 = ζi1; (3)

Random intercept αi and ζ ’s normally distributed
A maximum of T −1 V(ζit) variances can be identified

Empirical identification issue: Number of identifiable variances
depends on the data (correlations across time, # time points)
Default of all variances fixed at 1 is often reasonable

RI and AR1 Impact on Correlations Across Time (T=10)

(e) RI variance (R2) = 0.00 (f) RI variance (R2) = 0.25

(g) RI variance (R2) = 0.50 (h) RI variance (R2) = 0.75

Correlation Y∗
1 ,Y

∗
t = ψ +β t−1(1−ψ) where ψ is the random intercept variance, β is the

constant auto-regression among the residuals, and Y* variances are all 1

Identification Based on Correlations Across Time

Allowing time-varying means (probabilities Y=1), the thresholds are
identified by the proportions Y=1, while the rest of the model
parameters (RI variance, ARs, residual variances) are identified via the
correlations

Simulation example: T=10, RI variance = 1 (R-square of Y*≈0.5);
AR’s = 0.3, residual variances = 1. Corr (Y1,Yt):

0.5

0.6

0.7

0.8

0.9

1

2 3 4 5 6 7 8 9 10

The RI, AR, and residual variance parameter estimates are chosen to fit
the curve of the sample correlations at different time distances

Identification Based on Correlations Across Time Cont’d

0.5

0.7

0.8

0.9

1

2 3 4 5 6 7 8 9 10

The large T case:

With AR1 models and an autocorrelation of 0.3, the
autocorrelation source gives zero contribution at t ≈5
The correlations over longer time distances are solely due to the
random intercept variance
This implies that the random intercept variance is identifiable
Fixing a residual variance θ1 = 1 and knowing the random
intercept variance ψ , time 1 correlations identify the remaining
T-1 θ ’s: Corr (Y1,Yt) = ψ/(

√
ψ +1

√
ψ +θt)

The T-1 AR’s are then identified from among the remaining
correlations (T=5 has 10 correlations: 1 identifies ψ , 4 identify 4
θ ’s, and 4 identify the AR’s: 1 df)

More on Identification for a Binary Outcome
With T time points:

There are T sample proportions
T (T −1)/2 sample correlations

Example of a binary outcome with the RI-AR1 model and fixed
residual variances:

T parameters for the thresholds (unrestricted means), 1 parameter
for RI, T −1 parameters for AR1
With T = 3 there are 3+1+2 = 6 parameters = number of
proportions and correlations: Just-identified model
With T = 4 there are 4+1+3 = 8 parameters and 10 proportions
and correlations: 2 degrees of freedom
With T = 8 there are 8+1+7 = 16 parameters and 36 proportions
and correlations: 20 degrees of freedom

The first variance has different meaning than the other ones
because of no regression on previous time point

The first variance is identified but often is not significantly
different from the other variances which are fixed at 1
Depending on the parameter values, some residual variances can
also be identified but typically don’t change the fit of the model
while increasing SEs

Estimation

y1 y2 y3 y4

i

y5

0 0 0 0 0

y1* y2* y3* y4* y5*

i

y1 y2 y3 y4 y5y1* y2* y3* y4* y5*

y1*^ y2*^ y3*^ y4*^ y5*^

ML: Leads to too many dimensions of integration
WLSMV: Does not handle MAR; good with high coverage; fast
Bayes: Advantageous due to handling MAR

Bayes works poorly with the factor approach, but the residual
(hats) approach gives efficient computations - the ŷ∗it are not
unknowns like factors for which you draw values during the
MCMC iterations but are obtained as residuals by subtracting the
random intercept from Y* (Asparouhov & Muthén, 2022: RSEM)

Simulations for RI-AR1 with Binary Outcome
Time-varying thresholds and ARs
No Missing Data. WLSMV Estimation (Bayes results similar)
Parameter values:

Autocorrelations = 0.3
Random intercept variance = 1, residual variances = 1: R-square
due to random intercept = 0.5, R-square for residuals = 0.09
P(Y=1) = 0.20 (thresholds = 1.2)

Fixed variances (default):
T=3 gets good results for N=500
T=4 gets good results for N=500
T=8 gets good results for N=500 (no improvement due to no
time-invariant parameters)

Free 1st variance:
T=3 not identified
T=4 gets good results for N=2000
T=8 gets good results for N=500 (more corr’s for 1st var)

Free T-1 variances (often not a stable model in practice):
T=4 is not identified. T=5 gets good results for N=5000
T=8 gets good results for N=500

Section 7 Univariate Panel Data Models
for Binary Outcomes

Analyses of Data Set 2 Suicidal Ideation and
Substance Abuse

Panel Data Analysis: Model Notation Introduced in
Mplus Web Talk No. 4, Part 1

Dynamic models (y∗t regressed on y∗t−1):

AR: Auto-regressive, classic model which is dynamic by
definition. Used in CLPM
ARMA: Auto-regressive, Moving Average, classic model which
is dynamic by definition (can only be estimated by WLSMV)
D-RI-AR: AR of the classic, dynamic kind but with a random
intercept (RI) added
D-RI-ARMA: classic ARMA, that is, dynamic but with RI added
(can only be estimated by WLSMV)

RI with dynamic models for residuals (ytˆ regressed on yt−1ˆ):

RI-AR: AR is specified for the residual (“within-level”,
latent-variable centered) part. Used in RI-CLPM
RI-ARMA: ARMA is specified for the residual (“within-level”,
latent-variable centered) part (can only be estimated by WLSMV)

Average Tetrachoric Correlations Across Time

The plot suggest a lower asymptote larger than zero: Random
intercept variance (but much smaller than for the stress outcome
in the COMBINE data)

The plot suggests a higher AR and a higher random intercept
variance for Substance abuse than for Suicidal ideation

Univariate Analysis of Suicidal Ideation (N=737, T=8)

Outputs for all models are posted on the website of the talk

Frequency of response pattern with 0’s throughout = 155

4*28 = 112 bivariate cells. 5% = 6

Good fit, ok fit, marginal fit, poor fit

Model # par’s PPP/ # Significant Residuals Comment
χ2 Resp Pattern (obs. freq) Bivar

1. AR1 15 0.133 2 (155, 11) 22 Poor fit
2. AR2 21 0.378 0 3 OK fit

AR2* 21 χ2(15)=38 0 1 OK fit
(.0010)

3. ARMA11* 21 χ2(15)=17 0 0 Good fit
(.2972)

4. D-RI-AR1 16 0.474 0 0 Good fit
D-RI-AR1* 16 χ2(20)=32 0 0 Good fit

(.0395)
5. D-RI-ARMA11* 23 χ2(13)=11 0 0 Good fit

(.5734) Over-par’d.
Table continues

* denotes WLSMV

Univariate Analysis of Suicidal Ideation
Continued

Model # par’s PPP/ # Significant Residuals Comment
χ2 Resp Pattern (obs. freq) Bivar

6. RI-AR1 16 0.422 0 1 Good fit
RI-AR1* 16 χ2(20)=39 0 0 OK fit

(.0071)
7. RI-AR2 22 0.466 0 0 Good fit

RI-AR2* 22 χ2(14)=25 0 0 Good fit
(.0331)

8. RI-ARMA11* 22 χ2(14)=10 Negative V(iy). Over-par’d
(.7423)

9. RI-ARMA11* 21 χ2(15)=17 0 0 Good fit
V(iy)=0 (.2972) Same as 3.

* denotes WLSMV

Model 6 RI-AR1 Bayes Estimates for Suicidal Ideation

Substantial random
intercept variance =
0.756, S.E. (SD) = 0.102

STDYX estimates:

Estimate S.E.

I BY
Y1 0.656 0.025
Y2 0.642 0.026
Y3 0.653 0.026
Y4 0.636 0.032
Y5 0.632 0.032
Y6 0.603 0.038
Y7 0.607 0.034
Y8 0.581 0.035

Corr (Y1, Y8) due to RI =
0.656 × 0.581 = 0.381
(total est corr = 0.383)

R2 (Y8) = 0.5812 = 0.338

AR1 estimates
(standardized):

Estimate S.E.

Y2ˆ ON
Y1ˆ -0.231 0.131

Y3ˆ ON
Y2ˆ 0.035 0.117

Y4ˆ ON
Y3ˆ 0.304* 0.115

Y5ˆ ON
Y4ˆ 0.339* 0.101

Y6ˆ ON
Y5ˆ 0.488* 0.099

Y7ˆ ON
Y6ˆ 0.473* 0.086

Y8ˆ ON
Y7ˆ 0.565* 0.072

Univariate Analysis of Substance Use (N=737, T=8)

Frequency of response pattern with 0’s throughout = 201

Model # par’s PPP/ # Significant Residuals Comment
χ2 Resp Pattern (obs. freq) Bivar

1. AR1 15 0.259 1 (201) 14 Poor fit
2. AR2 21 0.462 0 0 Good fit

AR2* 21 χ2(15)=26 1(6) 0 OK fit
(.0339)

3. ARMA11* 21 χ2(15)=7 1 (6) 0 OK fit
(.9673) Over-par’d

4. D-RI-AR1 16 0.458 1 (201) 2 Marginal fit
D-RI-AR1* 16 χ2(20)=41 1 (6) 0 Poor fit

(.0032)
5. D-RI-AR2 22 0.482 1 (201) 1 Marginal fit

D-RI-AR2* 22 χ2(14)=27 1 (6) 0 OK fit
(.0184)

6. D-RI-ARMA* 23 χ2(13)=5 1 (6) 0 OK fit
(.9734) Over-par’d

Table continues

* denotes WLSMV

Univariate Analysis of Substance Use
Continued

Model # par’s PPP/ # Significant Residuals Comment
χ2 Resp Pattern (obs. freq) Bivar

7. RI-AR1 16 0.454 1 (201) 3 Marginal fit
RI-AR1* 16 χ2(20)=61 1 (6) 2 Poor fit

(.0000)
8. RI-AR2 22 0.508 1 (201)** 3 Marginal fit

RI-AR2* 22 χ2(14)=23 1 (6) 0 OK fit
(.0581)

9. RI-ARMA* 22 χ2(14)=16 1 (6) 0 OK fit
(.3272)

* denotes WLSMV
* z-score = 2.01

Section 8 Bivariate Binary Cross-Lagged Analyses
of Suicidal Ideation and Substance Abuse

The 20 Most Frequent Response Patterns (* = Missing)
Suicidal Ideation + Substance Abuse
Pattern Frequency Percentage

0000000000000000 104 14.1
0000000010000000 12 1.6
0000000001000000 10 1.4
*0000000*0000000 7 1.0
1100000000000000 7 1.0
0000000*0000000* 7 1.0
0010000000000000 6 0.8
0000000000000001 6 0.8
**000000**000000 6 0.8
0*0000000*000000 5 0.7
0000000000010000 5 0.7
0001000000000000 5 0.7
000000*0000000*0 5 0.7
000*0000000*0000 5 0.7
0000000000001000 5 0.7
0*******0******* 5 0.7
000*****000***** 4 0.5
0000000000000010 4 0.5
0000000000100000 4 0.5
0000000010001000 4 0.5

Percentage 0’s or missing: 28.8%

Bivariate Analysis: RI-CLPM (RI-AR11)

y1 y2 y3 y4 y5 y6 y7 y8

iy

z1 z2 z3 z4 z5 z6 z7 z8

iz

y1^ y2^ y3^ y4^ y5^ y6^ y7^ y8^

z1^ z2^ z3^ z4^ z5^ z6^ z7^ z8^

WLSMV Old Input Style for RI-CLPM (RI-AR11)

USEV = y1-y8 z1-z8;
CATEGORICAL = y1-z8;

ANALYSIS:
ESTIMATOR = WLSMV;
PARAMETERIZATION = THETA;
MODEL = NOCOV;

MODEL:
iy BY y1-y8@1;
iz BY z1-z8@1;
wy1 BY y1; wz1 BY z1;
wy2 BY y2; wz2 BY z2;
wy3 BY y3; wz3 BY z3;
wy4 BY y4; wz4 BY z4;
wy5 BY y5; wz5 BY z5;
wy6 BY y6; wz6 BY z6;
wy7 BY y7; wz7 BY z7;
wy8 BY y8; wz8 BY z8;

y1-z8@0;

wy1-wy8@1;
wz1-wz8@1;

wy2-wy8 PON wy1-wy7;
wz2-wz8 PON wz1-wz7;

wy2-wy8 PON wz1-wz7;
wz2-wz8 PON wy1-wy7;

wy1-wy8 PWITH wz1-wz8;

iy WITH iz;

WLSMV and Bayes New Input Style
for RI-CLPM (RI-AR11)

ANALYSIS:
ESTIMATOR = WLSMV;
PARAMETERIZATION = THETA;

MODEL:
iy BY y1-y8@1;
iz BY z1-z8@1;
y2ˆ-y8ˆ PON y1ˆ-y7ˆ;
z2ˆ-z8ˆ PON z1ˆ-z7ˆ;

y2ˆ-y8ˆ PON z1ˆ-z7ˆ;
z2ˆ-z8ˆ PON y1ˆ-y7ˆ;

y1-y8 PWITH z1-z8;
OUTPUT: STDYX RESIDUAL TECH10;
PLOT: TYPE = PLOT3;

ANALYSIS:
ESTIMATOR = BAYES;
BITERATIONS = (2000);
THIN = 10;
PROCESSORS = 8;

MODEL:
iy BY y1-y8@1;
iz BY z1-z8@1;
y2ˆ-y8ˆ PON y1ˆ-y7ˆ;
z2ˆ-z8ˆ PON z1ˆ-z7ˆ;

y2ˆ-y8ˆ PON z1ˆ-z7ˆ;
z2ˆ-z8ˆ PON y1ˆ-y7ˆ;

y1-y8 PWITH z1-z8;
OUTPUT: TECH8 STDYX RESIDUAL TECH10;
PLOT: TYPE = PLOT3;

Bivariate Analysis Results
for Suicidal Ideation and Substance Abuse

Outputs for all models are posted on the website of the talk

T=8 gives 16*(16-1)/2 = 120 bivariate tables, 480 cells, 5% = 24

Model # par’s PPP # Significant Residuals Comment
χ2 Resp Pattern Bivar

1. AR2 64 0.445 0 3 Good fit
AR2* 64 χ2(72)=115 0 1 Marginal fit
(CLPM2) (.0011)

2. ARMA* 64 χ2(72)=69 0 0 Good fit
(.5874) Over-par’d

3. D-RI-AR12 61 0.462 0 1 Good fit
D-RI-AR12* 61 χ2(75)=102 0 0 Good fit. Neg. V(iz)

(.0197)
4. RI-AR12 61 0.450 0 5 Good fit

RI-AR12* 61 χ2(75)=98 0 0 Good fit. Npd
(RI-CLPM12) (.0372)

5. RI-ARMA 65 χ2(71)=68 0 0 Good fit
Simple (.5940) Over-par’d
V(iy)=0

* denotes WLSMV

Testing of Time-Invariance Using Model Test (Wald Chi-2)
With Bayes Estimation of the RI-AR12 Model
Non-invariant model. Invariance testing of cross-lagged effects and covariances

ANALYSIS:
ESTIMATOR= BAYES;
BITERATIONS = (5000);
THIN = 10;
PROCESSORS = 8;

MODEL:
iy BY y1-y8@1;
iz BY z1-z8@1;
y2ˆ-y8ˆ PON y1ˆ-y7ˆ (a2-a8);
z2ˆ-z8ˆ PON z1ˆ-z7ˆ (b2-b8);
z3ˆ-z8ˆ PON z1ˆ-z6ˆ (c3-c8);

y2ˆ-y8ˆ PON z1ˆ-z7ˆ (d2-d8);
z2ˆ-z8ˆ PON y1ˆ-y7ˆ (e2-e8);

y1ˆ-y8ˆ PWITH z1ˆ-z8ˆ (f1-f8);
OUTPUT:

STANDARDIZED RESIDUAL
TECH8 TECH10 SVALUES;

PLOT:
TYPE = PLOT3;

MODEL TEST:
0 = d3-d2;
0 = d4-d2;
0 = d5-d2;
0 = d6-d2;
0 = d7-d2;
0 = d8-d2;
0 = e3-e2;
0 = e4-e2;
0 = e5-e2;
0 = e6-e2;
0 = e7-e2;
0 = e8-e2;
0 = f2-f1;
0 = f3-f1;
0 = f4-f1;
0 = f5-f1;
0 = f6-f1;
0 = f7-f1;
0 = f8-f1;

Results of Time-Invariance Testing with Model Test

WLSMV Bayes

Model χ2 test # Sig Residuals χ2/PPP # Sig Residuals

Invar (p-value) Patterns Bivar’s (p-value) Patterns Bivar’s

1. AR2 (CLPM2)
X-lags χ2(12)=24 χ2(12)=14

(.0195) (.3318)
X-lags + Covs χ2(19)=27 χ2(19)=18

(.1065) (.4970)
Both invar χ2(91)=130 0 2 0.458 0 6
(45 par’s) (.0043)

2. D-RI-AR12
X-lags χ2(12)=50 χ2(12)=14

(.0000) (.3258)
X-lags + Covs χ2(19)=54 χ2(19)=18

(.0000) (.5191)
Both invar Poor fit 0.446 0 3
(42 par’s)
No invar χ2(75)=102* 0 0 0.462 0 1
(61 par’s) (.0192)

* Negative V(iz)

Testing of Time-Invariance Continued

WLSMV Bayes

Model χ2 test # Sig Residuals χ2/PPP # Sig Residuals

Invar (p-value) Patterns Bivar’s (p-value) Patterns Bivar’s

3. RI-AR12
(RI-CLPM12)

X-lags χ2(12)=28 χ2(12)=17
(.0056) (.1561)

X-lags + Covs χ2(19)=31 χ2(19)=21
(.0375) (.3279)

Both invar χ2(94)=127* 0 2 0.426 0 9
(42 par’s) (.0129)

4. RI-ARMA
V(iy)=0

X-lags χ2(12)=25
(.0165)

X-lags + Covs χ2(19)=41
(.0023)

Both invar χ2(90)=109* 0 0
(46 par’s) (.0812)

* Negative V(iz)

Estimated Cross-Lagged Effects
Y: Suicidal Ideation, Z: Substance Abuse

WLSMV WLSMV BS = 500 Bayes
# Sig X-Lags # Sig X-Lags # Sig X-Lags

Model Cross-Lag Pos Neg NS Pos Neg NS Pos Neg NS

1. AR2 Invar. Y ON Z 1 0 0 1 0 0 1 0 0
(CLPM2) Z ON Y 1 0 0 1 0 0 1 0 0

2. D-RI-AR12 Y ON Z 7 0 0 5 0 2 1 0 6
Z ON Y 1 0 6 1 0 6 1 0 6

3. D-RI-AR12 Y ON Z 0 0 1
Invar. Z ON Y 0 0 1

4. RI-AR12 Yˆ ON Zˆ 3 0 4 2 0 5 1 0 6
Zˆ ON Yˆ 1 0 6 1 0 6 0 0 7

5. RI-AR12 Yˆ ON Zˆ 1 0 0 1 0 0 0 0 1
Invar. Zˆ ON Yˆ 0 0 1 0 0 1 0 0 1

6. RI-ARMA
V(iy)=0 fY ON fZ 1 0 0 1 0 0
Invar. fZ ON fY 0 0 1 1 0 0

Number of successful WLSMV bootstrap draws for the 5 models (not
M3): 459, 405, 93, 485, 470 (Asparouhov & Muthén, 2021)

Posterior Distribution of Bayes Estimate of Y7 ON Z6
for D-RI-AR12 Non-Invariant Model

Bayes posterior distribution has long tails due to uncertainty (small N)

WLSMV bootstrap distribution has similar shape (obtained by PLOT3)

Results Conclusions

Using Bayes for AR2 (CLPM2) and bootstrapped WLSMV RI-ARMA
with time-invariant cross-lags and residual covariances shows
significant but very small cross-lagged effects

Using Bayes with time-invariant cross-lags and residual covariances,
neither D-RI-AR12 nor RI-AR12 show significant cross-lagged
effects

Lack of power due to small N for low prevalence variables?
Lack of power of invariance testing?
Is there invariance across part of the time scale?

Where did the correlations between the two processes go?
- Intercept correlation and residual correlation

Covariates

y1 y2 y3 y4 y5 y6 y7 y8

iy

z1 z2 z3 z4 z5 z6 z7 z8

iz

y1^ y2^ y3^ y4^ y5^ y6^ y7^ y8^

z1^ z2^ z3^ z4^ z5^ z6^ z7^ z8^
x

Covariates: Gender, race, lunch (poverty indicator)
No effect on IY (suicidal ideation)

Using only the last 5 time points (T=5), female has a positive
effect and Black a negative effect

Female and Black have a negative effect on IZ (substance abuse)
Bengt Muthén RI-CLPM 78/ 162



Section 9 Planning Future Studies:
Power Estimation Using Monte Carlo Simulations

Increasing Power

How can power be increased to reject that cross-lagged effects
are zero?

Increase N
Reformulate variables to increase prevalence
Measure more often (e.g. Fall and Spring instead of annually)

Useful with time-invariance

Use multiple indicators and consider effects for latent constructs
How large does the sample need to be to make more
cross-lagged effects significant?

Quadrupling the sample size halves the SE’s
Monte Carlo simulations

Monte Carlo Simulation Studies

At which sample size N and number of time points T are the estimates
and standard errors trustworthy?

At which sample size and at which effect size can you detect an effect
if it exists? - What is the power? (Last column: % Sig Coeff.)

How much larger do N and T have to be to detect an effect when the
prevalence is lower? How much smaller can N and T be if prevalence
is higher?

How large do N and T have to be for model fit assessment to be
trustworthy?

How large do N, T, and effect sizes have to be for time-invariance
testing to be trustworthy?

How much larger do N and T have to be to distinguish between
competing models?

Muthén & Muthén (2002). How to use a Monte Carlo study to decide
on sample size and determine power. Structural Equation Modeling

Bayes Estimates for Non-Invariant D-RI-AR12

Est Post SD 2.5% 97.5% Est/SD EST/(SD/2)

Y2 ON
Y1 -0.118 0.104 0.131 -0.319 0.095
Z1 0.054 0.112 0.322 -0.169 0.273

Y3 ON
Y2 0.215 0.110 0.017 0.016 0.445 *
Z2 0.020 0.035 0.278 -0.047 0.088

Y4 ON
Y3 0.305 0.113 0.001 0.093 0.542 *
Z3 0.058 0.103 0.286 -0.136 0.261

Y5 ON
Y4 0.317 0.099 0.000 0.131 0.517 *
Z4 0.122 0.086 0.072 -0.043 0.301 1.42 2.84

Y6 ON
Y5 0.219 0.100 0.007 0.043 0.435 *
Z5 0.233 0.090 0.002 0.064 0.411 * 2.59 5.18

Y7 ON
Y6 0.209 0.100 0.016 0.016 0.415 *
Z6 0.147 0.082 0.035 -0.013 0.313 1.79 3.86

Y8 ON
Y7 0.413 0.122 0.000 0.192 0.667 *
Z7 0.089 0.086 0.145 -0.077 0.263

Simulation for Non-Invariant D-RI-AR12, N=737, Bayes
Pop. Est. Ave SD SE Ave MSE Cover %Sig

Y2 ON
Y1 -0.118 -0.1235 0.0852 0.0813 0.0073 0.926 0.334
Z1 0.054 0.0598 0.0896 0.0866 0.0081 0.942 0.136

Y3 ON
Y2 0.215 0.2279 0.0929 0.0920 0.0088 0.956 0.718
Z2 0.020 0.0183 0.0284 0.0297 0.0008 0.952 0.088

Y4 ON
Y3 0.305 0.3167 0.0954 0.0943 0.0092 0.960 0.960
Z3 0.058 0.0554 0.0853 0.0835 0.0073 0.938 0.102

Y5 ON
Y4 0.317 0.3254 0.0925 0.0873 0.0086 0.936 0.990
Z4 0.122 0.1225 0.0747 0.0731 0.0056 0.940 0.434

Y6 ON
Y5 0.219 0.2218 0.0819 0.0798 0.0067 0.938 0.826
Z5 0.233 0.2399 0.0779 0.0727 0.0061 0.932 0.932

Y7 ON
Y6 0.209 0.2128 0.0832 0.0839 0.0069 0.946 0.768
Z6 0.147 0.1508 0.0702 0.0683 0.0049 0.922 0.608

Y8 ON
Y7 0.413 0.4243 0.1087 0.1050 0.0119 0.942 0.992
Z7 0.089 0.0893 0.0761 0.0731 0.0058 0.944 0.266

SE’s lower than in real-data run due to real-data model not fitting perfectly

Simulation for Non-Invariant D-RI-AR12, N=3000, Bayes

Pop. Est. Ave SD SE Ave MSE Cover %Sig

Y2 ON
Y1 -0.118 -0.1184 0.0398 0.0397 0.0016 0.940 0.832
Z1 0.054 0.0546 0.0423 0.0421 0.0018 0.938 0.246

Y3 ON
Y2 0.215 0.2153 0.0449 0.0447 0.0020 0.956 0.996
Z2 0.020 0.0204 0.0138 0.0141 0.0002 0.944 0.302

Y4 ON
Y3 0.305 0.3102 0.0474 0.0457 0.0023 0.938 1.000
Z3 0.058 0.0564 0.0426 0.0410 0.0018 0.946 0.306

Y5 ON
Y4 0.317 0.3188 0.0414 0.0425 0.0017 0.944 1.000
Z4 0.122 0.1211 0.0349 0.0359 0.0012 0.962 0.934

Y6 ON
Y5 0.219 0.2194 0.0387 0.0390 0.0015 0.960 1.000
Z5 0.233 0.2351 0.0360 0.0357 0.0013 0.954 1.000*

Y7 ON
Y6 0.209 0.2114 0.0401 0.0410 0.0016 0.960 1.000
Z6 0.147 0.1477 0.0363 0.0336 0.0013 0.932 0.996

Y8 ON
Y7 0.413 0.4159 0.0525 0.0504 0.0028 0.942 1.000
Z7 0.089 0.0875 0.0366 0.0356 0.0013 0.930 0.684

Simulation for Non-Invariant D-RI-AR12, N=2000, Bayes

Pop. Est. Ave SD SE Ave MSE Cover %Sig

Y2 ON
Y1 -0.118 -0.1206 0.0495 0.0487 0.0024 0.950 0.664
Z1 0.054 0.0559 0.0525 0.0518 0.0027 0.954 0.180

Y3 ON
Y2 0.215 0.2187 0.0568 0.0550 0.0032 0.932 0.986
Z2 0.020 0.0200 0.0180 0.0174 0.0003 0.940 0.222

Y4 ON
Y3 0.305 0.3103 0.0577 0.0562 0.0033 0.940 1.000
Z3 0.058 0.0564 0.0503 0.0504 0.0025 0.956 0.202

Y5 ON
Y4 0.317 0.3179 0.0531 0.0521 0.0028 0.952 1.000
Z4 0.122 0.1220 0.0422 0.0440 0.0018 0.952 0.826

Y6 ON
Y5 0.219 0.2187 0.0477 0.0479 0.0023 0.950 0.996
Z5 0.233 0.2389 0.0443 0.0439 0.0020 0.946 1.000

Y7 ON
Y6 0.209 0.2112 0.0484 0.0502 0.0023 0.952 0.992
Z6 0.147 0.1481 0.0417 0.0412 0.0017 0.948 0.956

Y8 ON
Y7 0.413 0.4154 0.0644 0.0620 0.0042 0.934 1.000
Z7 0.089 0.0889 0.0443 0.0438 0.0020 0.946 0.532

Simulation for Non-Invariant RI-AR12, N=737, Bayes

Pop. Est. Ave SD SE Ave MSE Cover %Sig

Y2ˆ ON
Y1ˆ -0.113 -0.1352 0.1341 0.1311 0.0184 0.934 0.170
Z1ˆ -0.270 -0.2702 0.1609 0.1572 0.0258 0.946 0.412

Y3ˆ ON
Y2ˆ 0.027 0.0380 0.1067 0.1032 0.0115 0.950 0.068
Z2ˆ -0.056 -0.0725 0.0612 0.0656 0.0040 0.950 0.212

Y4ˆ ON
Y3ˆ 0.265 0.2651 0.1066 0.1123 0.0113 0.960 0.666
Z3ˆ -0.037 -0.0361 0.1028 0.1045 0.0105 0.948 0.070

Y5ˆ ON
Y4ˆ 0.242 0.2351 0.1093 0.1045 0.0120 0.940 0.614
Z4ˆ 0.165 0.1686 0.1126 0.1140 0.0127 0.956 0.336

Y6ˆ ON
Y5ˆ 0.404 0.3964 0.1154 0.1105 0.0133 0.940 0.934
Z5ˆ 0.314 0.3254 0.1264 0.1219 0.0161 0.948 0.802*

Y7ˆ ON
Y6ˆ 0.402 0.3948 0.1080 0.1071 0.0117 0.936 0.948
Z6ˆ 0.050 0.0589 0.1240 0.1279 0.0154 0.958 0.080

Y8ˆ ON
Y7ˆ 0.542 0.5516 0.1444 0.1424 0.0209 0.940 0.958
Z7ˆ 0.116 0.1111 0.1370 0.1355 0.0188 0.936 0.156

Bengt Muthén RI-CLPM 86/ 162



Simulation for Non-Invariant RI-AR12, N=3000, Bayes

Pop. Est. Ave SD SE Ave MSE Cover %Sig

Y2ˆ ON
Y1ˆ -0.113 -0.1139 0.0610 0.0594 0.0037 0.946 0.488
Z1ˆ -0.270 -0.2748 0.0759 0.0726 0.0058 0.936 0.962

Y3ˆ ON
Y2ˆ 0.027 0.0285 0.0492 0.0502 0.0024 0.950 0.086
Z2ˆ -0.056 -0.0565 0.0294 0.0295 0.0009 0.944 0.540

Y4ˆ ON
Y3ˆ 0.265 0.2664 0.0526 0.0540 0.0028 0.948 0.998
Z3ˆ -0.037 -0.0373 0.0502 0.0481 0.0025 0.920 0.134

Y5ˆ ON
Y4ˆ 0.242 0.2416 0.0493 0.0503 0.0024 0.960 0.998
Z4ˆ 0.165 0.1652 0.0542 0.0546 0.0029 0.952 0.854

Y6ˆ ON
Y5ˆ 0.404 0.4041 0.0503 0.0525 0.0025 0.960 1.000
Z5ˆ 0.314 0.3126 0.0598 0.0578 0.0036 0.934 1.000*

Y7ˆ ON
Y6ˆ 0.402 0.4013 0.0497 0.0502 0.0025 0.956 1.000
Z6ˆ 0.050 0.0506 0.0595 0.0600 0.0035 0.956 0.122

Y8ˆ ON
Y7ˆ 0.542 0.5450 0.0634 0.0630 0.0040 0.938 1.000
Z7ˆ 0.116 0.1129 0.0593 0.0608 0.0035 0.954 0.458

Power to Reject Time Invariance: Wald Test Simulations
Results for CLPM2 (AR2) WLSMV and RI-CLPM Bayes

Real data: Chi-square (19) = 27 (.1065)

Simulated data (N=737):

Degrees of freedom 19

Mean 59.530
Std Dev 13.506
Number of successful computations 472

Proportions
Expected Observed

0.990 1.000
0.980 1.000
0.950 1.000
0.900 1.000
0.800 1.000
0.700 1.000
0.500 1.000
0.300 1.000
0.200 0.998
0.100 0.994
0.050 0.994
0.020 0.977
0.010 0.977

Real data: Chi-square (19) = 21 (.3279)

Simulated data (N=737):

Degrees of freedom 19

Mean 46.463
Std Dev 11.225
Number of successful computations 500

Proportions
Expected Observed

0.990 1.000
0.980 1.000
0.950 1.000
0.900 1.000
0.800 1.000
0.700 1.000
0.500 0.998
0.300 0.990
0.200 0.988
0.100 0.962
0.050 0.938
0.020 0.888
0.010 0.820

Wald Test Simulation Conclusions

Is the power to reject time invariance in the real data too low due to
sample size and prevalence not being large enough?

The simulations show that power to reject time invariance is high
assuming the model is correct (which it is in the simulations)

The real-data analysis has larger SEs due to model not being exactly
correct. It therefore gets lower power

The power to reject time invariance may be low in this case

Section 10 Other Analysis Approaches:
Measurement Error

Observed Y as Predictor Instead of Latent Y*
Reciprocal Modeling

Measurement Error

Measurement error with various restrictions can be added to CLPM and
RI-CLPM for categorical outcomes and works ok in simulations, but

For the Suicide-Substance data, adding measurement error fits
better for some of the models but fails for several other models
Time-invariant measurement error does not give time-invariant
reliability due to Y* variance changing over time (parameter
constraints needed for imposing time-invariant reliability)

See also MEAR in the Asparouhov & Muthén (2022) RSEM paper

Difficult in practice even for continuous outcomes - avoid

Observed Y Instead of Latent Y*

y1 y2 y3 y4

iy

y5

iz

z1 z2 z3 z4 z5

Random intercept, but no separation of between and within:
D-RI-AR1 model for observed binary outcomes, not Y*
Mplus Analysis option Predictor = Observed using Bayes
(WLSMV not available). Bayes PPP not available
Predictor = Observed not relevant for RI models with hats
variables which are by definition latent

Related econometric modeling:
Individual-specific heterogeneity; individual effects
Honoré & Kyriazidou (2019). Panel vector autoregressions with
binary data. In Panel Data Econometrics (edited book)

Univariate Results for Observed Y vs Latent Y* (Bayes)
Suicidal Ideation and Substance Abuse

# Significant Residuals

Model # par’s PPP Resp. patterns (freq.) Bivar

Suicidal Ideation
AR1 Observed 15 2 (155,11) 46

Latent 15 0.133 2 (155,11) 22

AR2 Observed 21 2 (155,11) 12
Latent 21 0.378 0 3

D-RI-AR1 Observed 16 0 0
Latent 16 0.474 0 0

Substance Abuse
AR1 Observed 15 2 (201,6) 49

Latent 15 0.259 1 (201) 14

AR2 Observed 21 1 (201) 10
Latent 21 0.462 0 1

D-RI-AR1 Observed 16 0 3
Latent 16 0.458 1 (201) 2

D-RI-AR2 Observed 22 0 0
Latent 22 0.482 1 (201) 1

Most frequent pattern of 0’s: 155, 201. Bivar 5%5 = 6

Bivariate Results for Observed Y vs Latent Y* (Bayes)
Suicidal Ideation and Substance Abuse

# Significant Residuals

Model # par’s PPP Resp. patterns (freq.) Bivar

D-RI-AR11 Observed 55 0 5
Latent 55 0.486 0 2

D-RI-AR12 Observed 61 0 2
Latent 61 0.462 0 1

D-RI-AR12 cross-lagged estimates:

Observed:

Y6 ON Z5 positive and significant
Z4 ON Y3 positive and significant
Z7 ON Y6 positive and significant

Latent:

Y6 ON Z5 positive and significant
Z4 ON Y3 positive and significant

Lagged and Contemporanous Influence for Y*

The bottom model with reciprocal interaction may be more
realistic but is not identified without parameter restrictions

Identification difficulties are described in Greenberg & Kessler
(1982)

Lagged versus Contemporaneous Influence for Y*

Regular Reciprocal

Cross-lagged effects suitable for variables referring to current status

Reciprocal effects may be suitable for variables referring to past status
(e.g., suicidal ideation and substance abuse during the last year)

Cross-lagged effects may be of ignorable magnitude
Reciprocal model is identified because each DV has its own
predictor (old econometric rule)

Cross-Lagged versus Reciprocal Model for Y*

The cross-lagged and reciprocal models have the same number
of parameters

For continuous variables, the two models fit the same. But a test
of time-invariant cross-lagged versus reciprocal effects gives
different results

For categorical variables, the two models fit differently

Using the cross-lagged model to analyze data generated by the
reciprocal model will show cross-lagged effects: Seeing
cross-lagged effects does not rule out the reciprocal model

Reciprocal Model Results for Suicide-Substance Data

Simulated data shows good performance of the model for large
samples sizes (N of the order of at least 2,000)

Reciprocal AR model does not fit the Suicide-Substance data

Reciprocal RI-AR fits well but does not give believeable results
for the reciprocal effects and their residual covariances with
some very high correlations among estimates

The model using Predictor = Observed is not identified
(Maddala, 1983, Section 5.7)

Hybrid Model: Lagged and Contemporaneous Effects

A hybrid model is identified and may be useful if there is strong
substantive reason for the specification

Bengt Muthén RI-CLPM 99/ 162



Section 11 Latent Transition Analysis:
Mover-Stayer LTA

RI-LTA
Bivariate LTA

Studying Changes Over Time
Using Latent Transition Analysis (LTA)

Cross-lagged panel modeling needs outcomes that show change
over time

LTA focuses on the changes over time

Mixture modeling

Maximum-likelihood estimation

For an introduction to LTA, see Mplus Web Talk No. 2 at
https://www.statmodel.com/Webtalk2.shtml

LTA (Hidden Markov Model)

c3

y1 y2 y3 y4 y5 y6 y7 y8

c1 c2 c4 c5 c6 c7 c8

Transitions over time
Measurement error
3 versions:

Regular LTA (2 classes per time point, lag-1, non-stationary,
28 = 256 class combinations)
Mover-Stayer (adding an M-S class gives 29 = 512 class
combinations)
Random intercept (RI-LTA), Muthén & Asparouhov (2022)
https://www.statmodel.com/download/
MuthenFINAL.pdf

Mover-Stayer LTA (Hidden Markov Model)

c3

y1 y2 y3 y4 y5 y6 y7 y8

c1 c2 c4 c5 c6 c7 c8

c

An example of Stayers: Individuals who have 0’s at all time
points

RI-LTA

c3

y1 y2 y3 y4 y5 y6 y7 y8

c1 c2 c4 c5 c6 c7 c8

i

Also possible: Mover-Stayer RI-LTA

LTA MLR Results for Suicidal Ideation (N=737, T = 8)

Outputs are posted on the website of the talk

No χ2 test of model fit available for LTA (no Y*’s)

Comparison of model fit with previous models can be made using
the standardized residuals

# Sig Standardized Residuals

Model # par’s LL BIC Response Patterns Bivariate Comments

1. LTA 2c 17 -2150 4411 2 (155, 6) 6 Poor fit

2. LTA 2c 23 -2139 4430 2 (155, 6) 2 Poor fit
Lag 2

3. LTA 2c 19 -2138 4401 1 (6) 3 OK fit
Mover-Stayer 36% stayers

4. RI-LTA 2c 18 -2131 4380 1 (6) 3 OK fit

5. RI-LTA 20 -2131 4394 1 (6) 3 OK fit
Mover-Stayer 5% stayers

RI-LTA 2c gives 84% in low class, 16% in high class

RI-LTA Estimates for Suicidal Ideation

Measurement Probabilities
Latent Class
Low High

y=0 0.87 0.21
y=1 0.13 0.79

Estimated probabilities of y = 1, where P(Y=1) =
P(C=High)*P(Y=1 | C=High) + P(C=Low)*P(Y=1 | C=Low)

0.00

0.05

0.10

0.15

0.20

0.25

19 20 21 22 23 24 25 26

Age

Estimated Y=1
Probabilities

Latent Class 2
Probabilities

Values for top curve obtained from Tech10, bottom curve from Tech15

RI-LTA EstimatedTransition Probabilities For Suicide
C1 → C2

1 2
1 0.914 0.086
2 1.000 0.000

C2 → C3
1 2

1 0.966 0.034
2 1.000 0.000

C3 → C4
1 2

1 0.926 0.074
2 0.008 0.992

C4 → C5
1 2

1 0.929 0.071
2 0.268 0.732

C5 → C6
1 2

1 0.968 0.032
2 0.202 0.798

C6 → C7
1 2

1 0.928 0.072
2 0.299 0.701

C7 → C8
1 2

1 0.985 0.015
2 0.258 0.742

First 3 transition tables different from the rest. Invariance for the last 5 time
points?

LTA MLR Results for Substance Abuse (T = 8)

# Sig Standardized Residuals

Model # par’s LL BIC Response Patterns Bivariate Comments

1. LTA 2c 17 -1804 3721 2 (201, 6) 10 Poor fit

2. LTA 2c 23 -1785 3722 2 (201, 6) 1 Poor fit
Lag 2

3. LTA 2c 19 -1784 3695 0 0 Good fit
Mover-Stayer 58% stayers

4. RI-LTA 2c 18 -1783 3685 0 0 Good fit
Poor meas.

5. RI-LTA 20 -1783 3699 0 0 Good fit
Mover-Stayer 5% stayers

LTA Mover-Stayer Estimates for Substance Abuse

Measurement Probabilities
Latent Class
Low High

y=0 0.98 0.12
y=1 0.02 0.88

Estimated probabilities of y = 1, where P(Y=1) =
P(C=High)*P(Y=1 | C=High) + P(C=Low)*P(Y=1 | C=Low)

0.00

0.05

Age

Estimated Y=1
Probabilities

Latent Class 2
Probabilities

LTA Mover-Stayer Estimated Transition Probabilties
For Substance Abuse Movers Obtained from TECH15

C1 → C2
1 2

1 1.00 0.00
2 0.00 1.00

C2 → C3
1 2

1 0.72 0.28
2 0.71 0.29

C3 → C4
1 2

1 0.62 0.38
2 0.53 0.47

C4 → C5
1 2

1 0.78 0.22
2 0.45 0.55

C5 → C6
1 2

1 0.87 0.14
2 0.50 0.50

C6 → C7
1 2

1 0.88 0.12
2 0.39 0.61

C7 → C8
1 2

1 0.90 0.10
2 0.54 0.46

Bivariate LTA

y1 y5y4y3y2

z1 z2 z3 z4 z5

cz1 cz2 cz3 cz4 cz5

cy1 cy5cy4cy3cy2

Related bivariate LTA:
ALTA (Associate Latent Transition Analysis): Flaherty (2008),
Witkiewitz & Villarroel (2009), Bray et, al (2010)

Mover-Stayer bivariate LTA with variations of the M-S latent
class variable:

For each outcome, one for both, for only one outcome

Bivariate RI-LTA

Bivariate LTA: Computational Considerations

Bivariate LTA is time and memory demanding for high T even
with only 2 classes due to proliferation of latent class
combinations:

Regular LTA with T = 7 results in 214 = 16,384 combination
classes which is slow but doable on a good computer
Mover-Stayer LTA adds a latent class variable for each outcome
which for T = 7 results in 2×2×214 = 216 = 65,536
If 16,384 is the max, 2×2×2x = 16,384 gives x = (ln 16,384 - ln
4)/ln 2 = 12, that is T = 12/2 = 6
The T = 8 analyses show more stable development after the first 3
occasions, suggesting analysis of the last 5 (T = 5), which results
in 2×2×210 = 4,096

Univariate LTA MLR Results for T = 5
Imposing Stationarity (Time-Invariant Transitions)

Suicidal Ideation

# Sig Standardized Residuals

Model # par’s LL BIC Response Patterns Bivariate Comments

1. LTA 2c 11 -1389 2851 2 (7, 7) 0 OK fit

2. LTA 2c 13 -1383 2852 1 (7) 0 Good fit
Mover-Stayer 36% stayers

3. RI-LTA 2c 12 -1383 2844 1 (7) 0 Good fit
Poor Meas.

Substance Abuse

1. LTA 2c 11 -1142 2356 1(6) 0 Good fit

2. LTA 2c 13 -1142 2370 0 0 Good fit
Mover-Stayer 58% stayers

3. RI-LTA 12 -1140 2360 0 0 Good fit
Poor meas.

Bivariate LTA MLR Results for T = 5
Imposing Stationarity (Time-Invariant Transitions)

Outputs are posted on the website of the talk

Most frequent pattern of all 0’s is observed for 193 individuals

# Sig Standardized Residuals

Model # par’s LL BIC Response Patterns Bivariate Comments

1. LTA 2c 13 -2497 5080 5 (193, 7, 7, 6, 6) 1 Marginal fit

2. LTA 2c 18 -2488 5094 3 (193, 7, 6) 3 Marginal fit
Mover-Stayer 66 % stayers (Y)

for both 45 % stayers (Z)

Cross-Lagged Effects for LTA Using Model 1
in Logit and OR Terms

CY ON CZ = 2.172 (SE = 0.651). OR = exp (2.172) = 8.77
Translating logit CI to OR CI by exponentiation: [2.45 31.14]

Significant effect

CZ ON CY = 0.471 (SE = 0.528). OR = exp(0.471) = 1.602
Translating logit CI to OR CI by exponentiation: [0.57 4.51]

Non-significant effect

Conclusion: Substance abuse influences suicidal ideation, not the
other way around

The model is analogous to bivariate AR1 (CLPM) with
time-invariant cross-lagged effects - but adding measurement
error

The Mover-Stayer Model 2 results also point to significant CY
ON CZ and insignificant CZ ON CY (among Movers)

Further Detail: Model Test of Probability Differences
Results in Probability terms obtained by Model Constraint:

1 P(CYt=1 | CYt-1=1, CZt-1=1) = 0.980
2 P(CYt=1 | CYt-1=1, CZt-1=2) = 0.851
3 P(CYt=1 | CYt-1=2, CZt-1=1) = 0.090
4 P(CYt=1 | CYt-1=2, CZt-1=2) = 0.011

1 - 2: Effect of CZ on CY for the low CY class (CYt-1=1)
The probability of staying in the low CY class is decreased from
0.980 to 0.851 by previously being in the high compared to the
low CZ class

3 - 4: Effect of CZ on CY for the high CY class (CYt-1=2)
The probability of transitioning from the high to the low CY class
is decreased from 0.090 to 0.011 by previously being in the high
compared to the low CZ class

Testing both the 1 -2 and the 3 - 4 differences using Model Test:
Chi-2 (2) = 7.2 (.0280)
Testing first difference: Chi-2 (1) = 6.3 (.0124)
Testing second difference: Chi-2 (1) = 2.8 (.0937)

Bengt Muthén RI-CLPM 116/ 162



Other LTA Models

LTA also possible for ordinal variables (and nominal variables)
5-category alcohol risk outcome for T = 5 analyzed with up to 5
classes with regular LTA and up to 4 classes with Mover-Stayer
LTA and RI-LTA

How do you bring the two-part ordinal parts into the LTA?

How do you combine an LTA for alcohol risk with a model for
the continuous stress variable to capture cross-lagged effects?

Section 12 Ordered Categorical (Ordinal)
Outcomes

Two-Part Ordinal Model

Bengt Muthén RI-CLPM 118/ 162



Ordered Categorical (Ordinal) Outcomes

Data set 3: 5-category negative affect item Irritability

Question format: Today I felt... (1 = Not at all, 2 = A little, 3 =
Moderately, 4 = Quite a bit, 5 = Extremely)

Analysis of days 1-10

N = 271

Average Correlations as a Function of the Time Distance
for 5-Category Irritability Item (N=271, T=10)

The plot suggests a low AR and a high random intercept variance

Univariate Analysis of 5-Categ. Irritability (N=271, T=10)

Number of subjects with 0’s throughout = 52 (19%). This is the most
frequent response pattern

Total number of bivariate cells: 5*5*10(10-1)/2 = 1125. 5% = 56

The unrestricted multivariate probit model has 10*4 thresholds and
10(10-1)/2 = 45 correlations: 85 parameters

Model # par’s PPP/ # Significant Residuals Comment
χ2 Resp Pattern (obs. freq) Bivar

1. Unrestr. 85 0.498 3 (2,1,1) 35 Good fit
2. AR1 49 0.002 4 (52, 2, 1, 1) 121 Poor fit
3. RI-AR1 50 0.382 3 (2, 1, 1) 32 Good fit

RI-AR1* 50 χ2(35)=64 3 (2, 1, 1) 49 Marginal fit
(.0018)

4. RI-ARMA11* 58 χ2(27)=35 3 (2, 1, 1) 47 Good fit
(.1295)

* denotes WLSMV

Models 3 and 4 both have sizeable random intercept variance - the
random intercept R-square for RI-ARMA Y*’s is about 0.5

Only 3 of 9 ARs are significant, agreeing with correlation plot

Data Set 1: Analysis of 5-Category Alcohol Risk
and 8-Category Heavy Drinking (N=1375, T=8)

5-category alcohol risk:
50%

25%

9% 9%
7%

Abstinence Low Risk Medium Risk High Risk Very High Risk

Most frequent pattern is 0’s throughout with frequency 312 (23 %)

Total number of bivariate cells = 25*8(8-1)/2 = 700. 5% = 35

The unrestricted multivariate probit model has 8*4 thresholds and
8(8-1)/2 = 28 correlations: 60 parameters

Number of significant response pattern residuals: 5 for patterns
with observed frequencies 312 (estimated as 267), 46, 12, 6, 6
Number of significant bivariate cells = 273 which is 39% of all
bivariate cells

Poor fit: A new model is needed

Bivariate Models for Ordered Categorical (Ordinal) Variables

59

St
ro

ng
ly

Ag
re

e

y j*
yj

St
ro

ng
ly

D
is

ag
re

e

Strongly
Disagree

Strongly
Agree

y k*

117

yk

1 The probit model for 2 variables in the figure has C1-1+C2-1+1
(thresholds+1 polychoric corr) = C1+C2-1 parameters where C is the
number of categories: 9 for C=5

2 The unrestricted multinomial model for 2 variables has C1*C2-1
parameters: 24 for C=5. This is the model tested against in the
TECH10 bivariate tests

3 Intermediate models are possible. E.g. a two-part ordinal model for 2
variables has C1+C2 parameters: 10 for C=5 (1 more than probit)

Two-Part Ordinal Model

Two-part growth modeling with continuous outcomes: Olsen &
Schafer (2001), JASA; two-part GMM - Muthén (2001)

Two-part regression analysis with an ordinal outcome: Muthén,
Muthén & Asparouhov (2016). RMA book, page 292

Two-part ordinal multivariate probit model: Version 8.8 TECH10
With two C-category ordinal variables there are C1-1+C2-1
thresholds and two correlations: a correlation between the binary
parts as well as the ordinal parts (C1-1+C2-1+2 = C1+C2)

Mplus Input for Univariate (Single Process, T=8)
Unrestricted Two-Part Ordinal Model Using Bayes

USEVARIABLES = u1-u8 p1-p8;
CATEGORICAL = u1-u8 p1-p8;

DATA TWOPART:
NAMES = z1-z8; ! from Names list
BINARY = u1-u8;
CONTINUOUS = p1-p8;
CUTPOINT = 0;
TRANSFORM = NONE;

ANALYSIS:
ESTIMATOR = BAYES;
BITERATIONS = (10000);
THIN = 10;
PROCESSORS = 8;

MODEL:
u1-u8 WITH u1-u8;
p1-p8 WITH p1-p8;
! Approach that keeps ut and pt
! uncorrelated in line with
! Asparouhov-Muthén (2022)
! Residual Structural Equation
! Models, Section 3.5:
p1ˆ-p8ˆ ON u1ˆ-u8ˆ;
p1ˆ-p8ˆ PON u1ˆ-u8ˆ@0;

OUTPUT:
STANDARDIZED RESIDUAL
TECH8 TECH10;

PLOT:
TYPE = PLOT3;

Correlations among 16 variables (binary and ordinal parts), where 8 are
fixed at zero (u-p correlations are not well identified since p is not
observed when u=0). 2T*(2T-1)/2 - T: 112

For each variable, there is one threshold for the binary part of and C-2
thresholds for the ordinal part. T + T*(C-2): 32

Mplus Input for Univariate (Single Process)
RI-AR2 Two-Part Ordinal Model Using Bayes

USEVARIABLES = u1-u8 p1-p8;
CATEGORICAL = u1-u8 p1-p8;

DATA TWOPART:
NAMES = z1-z8;
BINARY = u1-u8;
CONTINUOUS = p1-p8;
CUTPOINT = 0;
TRANSFORM = NONE;

ANALYSIS:
ESTIMATOR = BAYES;
BITERATIONS = (10000);
THIN = 10;
PROCESSORS = 8;

MODEL:
ib BY u1-u8@1;
ip BY p1-p8@1;
u2ˆ-u8ˆ PON u1ˆ-u7ˆ;
u3ˆ-u8ˆ PON u1ˆ-u6ˆ;
p2ˆ-p8ˆ PON p1ˆ-p7ˆ;
p3ˆ-p8ˆ PON p1ˆ-p6ˆ;
! u1ˆ-u5ˆ PWITH p1ˆ-p5ˆ;

OUTPUT:
STANDARDIZED RESIDUAL
TECH8 TECH10;

PLOT:
TYPE = PLOT3;

The two parts are correlated only via the correlation between the two
random intercepts. The residual correlations are not well identified
since p is not observed when u=0

Bengt Muthén RI-CLPM 126/ 162



Analysis of 5-Category Alcohol Risk Using
Regular and Two-Part Ordinal Models (N=1375, T=8)

Model # par’s PPP/ # Significant Residuals Comment
χ2 Resp Pattern (obs. freq) Bivar

Regular ordinal probit

1. Unrestricted 60 0.498 5 (312) 273 (39%) Poor fit
2. AR2 45 0.151 5 (312) 277 Poor fit
3. RI-AR2 46 0.135 5 (312) 274 Poor fit
4. RI-ARMA11* 46 χ2 (14)=39 5 (312) 279 Poor fit

Two-part ordinal probit

5. Unrestricted 144 0.472 1 (12) 29 (4%) Good fit
6. AR2 58 0.145 2 (46, 13) 106 (15%) Poor fit
7. RI-AR2 61 0.228 1 (12) 52 (7%) OK fit
8. RI-ARMA11† NA

* denotes WLSMV. † No WLSMV for two-part

Model 7 (RI-AR2) has sizeable variances for the two random intercepts
- the random intercept R-squares range from 0.3 to 0.6 with slightly
higher values for the binary part

Regular ordinal probit Model 3 (RI-AR2) has ignorable random
intercept variance (est. less than s.e.) and R-squares about 0.04

Analysis of 8-Category Heavy Drinking
Using Two-Part Ordinal Models (N=1375, T=8)

66%

11%
6% 6%

76543210

4% 2% 2%3%

Number of Heavy Drinking Days Per Week

Most frequent pattern is 0’s throughout with frequency 503 (37 %)

Total number of bivariate cells = 64*8(8-1)/2 = 1792. 5% = 90

Analysis of 8-Category Heavy Drinking Using
Regular and Two-Part Ordinal Models (N=1375, T=8)

Model # par’s PPP/ # Significant Residuals Comment
χ2 Resp Pattern (obs. freq) Bivar

Regular ordinal probit

1. Unrestricted 84 0.492 3 (38, 8, 7) 211 (12%) Poor fit
2. AR2 69 0.233 3 (503 7, 5) 228 Poor fit
3. RI-AR2 70 0.230 3 (503, 7, 5) 229 Poor fit
4. RI-ARMA11* 70 χ2 (14)=30 4 (7,7,6,5) 225 (13%) Poor fit

Two-part ordinal probit

5. Unrestricted 168 0.454 1 (7) 45 (3%) Good fit
6. AR2 82 0.053 6 (503) 124 (7%) Poor fit
7. RI-AR2 85 0.165 3 (8, 7, 5) 70 (4%) OK fit
8. RI-ARMA11† NA

* denotes WLSMV. † No WLSMV for two-part

Model 7 (RI-AR2) has sizeable variances for the two random intercepts
- the random intercept R-squares range from 0.4 to 0.6 with somewhat
higher values for the binary part

Regular ordinal probit Model 3 (RI-AR2) has ignorable random
intercept variance (est. less than s.e.) and R-squares about 0.04

Bengt Muthén RI-CLPM 129/ 162



Bivariate Analysis of Stress and Alcohol Risk
Week 1 - Week 16 (N=1375, T=8)

Returning to the question in the introduction: Does stress
influence alcohol risk or the other way around?

Stress causes drinking (Armeli et al., 2000 in J of Personality
and Social Psych)

The alcohol treatment setting may produce a different picture

Bivariate Analysis of Stress and Alcohol Risk
Week 1 - Week 16 (N=1375, T=8)

Average Correlations as a Function of the Time Distance

Two-part ordinal RI-AR2 model chosen for 5-category alcohol risk

RI-AR2 model chosen for stress

RI-ARMA model for stress fits better but cannot be combined
with two-part ordinal modeling because RI-ARMA requires
WLSMV which does not handle two-part (not MAR)

Mplus Input for Bivariate Analysis of Stress
and Alcohol Risk Using RI-AR2 and Two-Part Ordinal

USEVARIABLES = y1-y8 u1-u8 p1-p8;
CATEGORICAL = u1-u8 p1-p8;

DATA TWOPART:
NAMES = z1-z8;
BINARY = u1-u8;
CONTINUOUS = p1-p8;
CUTPOINT = 0;
TRANSFORM = NONE;

ANALYSIS:
ESTIMATOR = BAYES;
BITERATIONS = (10000);
THIN = 10;
PROCESSORS = 8;

MODEL:
iy BY y1-y8@1;
ib BY u1-u8@1;
ip BY p1-p8@1;

! univariates:
y2ˆ-y8ˆ PON y1ˆ-y7ˆ;
y3ˆ-y8ˆ PON y1ˆ-y6ˆ;
u2ˆ-u8ˆ PON u1ˆ-u7ˆ;
u3ˆ-u8ˆ PON u1ˆ-u6ˆ;
p2ˆ-p8ˆ PON p1ˆ-p7ˆ;
p3ˆ-p8ˆ PON p1ˆ-p6ˆ;
! bivariates:
y2ˆ-y8ˆ PON u1ˆ-u7ˆ;
y2ˆ-y8ˆ PON p1ˆ-p7ˆ;
u2ˆ-u8ˆ PON y1ˆ-y7ˆ;
p2ˆ-p8ˆ PON y1ˆ-y7ˆ;

OUTPUT:
STANDARDIZED RESIDUAL
TECH8 TECH10;

PLOT:
TYPE = PLOT3;

Results for Bivariate Analysis of Stress
and Alcohol Risk Using RI-AR2 and Two-Part Ordinal

Stress = Y, Alcohol risk = Z

#par.’s = 121, PPP = 0.054, # signif. response pattern residuals =
1 (freq. = 12), #significant bivariate residuals = 55
# significant cross-lagged effects:

Y ON binary part of Z: 4 out of 7, stdyx = 0.12 - 0.17
Y ON ordinal part of Z: 7 out of 7, stdyx = 0.16 - 0.21
Binary part of Z ON Y: 0
Ordinal part of Z ON Y: 0

Increased alcohol risk has a significant effect on increase of
stress, not the other way around

Abstinence or not has less of an effect on stress than higher
degree of alcohol risk (non-abstinence)

Section 13 Allowing for a Trend:
Latent Growth Analysis

Longitudinal LCA
Latent Class Growth Analysis

Growth Mixture Modeling

Bengt Muthén RI-CLPM 134/ 162



COMBINE Stress and Alcohol Risk: Baseline - Week 16

Stress
Means

Can a left-out random effect capturing a trend distort RI or AR? Would
cross-lagged effects be affected?

Allowing for a Trend: Latent Growth Analysis (Cont’s Y)
Adding linear and quadratic random effects to the random
intercept also contributes to correlations across time:

yti =  η0i + η1i xt + εti ,
xt = 0, 1, …, T – 1. 

Cov(yti ,yt ́i ) = V(η0i) + V(η1i) xt xt ́
+ Cov(η0i , η1i) (xt + xt ́)
+ Cov(εti , εt ́i  ).

t

η0 η1

V(η1i ) xt xt   :

t

t

η0 η1

Cov(η0i , η1i) xt :

t t

η0 η1

Analysis of Stress: Comparing Model Fit
for RI-AR2 vs Linear Growth AR2 Using MLR

Week 1 - Week 16 (N=1375, T=8)

Model # par’s LL BIC Chi-square RMSEA CFI

RI-AR2 30 -19928 40073 χ2(14)=33 0.032 0.995
(.0027) (<.05=.983)

Linear Growth 26 -19935 40057 χ2(18)=46 0.034 0.993
with AR2 (.0003) (<.05=.987)

Input for linear growth AR2 model:

Model:
! Weeks: 1, 2, 4, 6, 8, 10, 12, 16
i s | y1@0 y2@.1 y3@.3 y4@.5 y5@.7 y6@.9 y7@.1.1 y8@.1.5;
! Auto-regression among residuals - new in version 8.7:
y2ˆ-y8ˆ PON y1ˆ-y7ˆ;
y3ˆ-y8ˆ PON y1ˆ-y6ˆ;

Analysis of Stress: Comparing MLR Estimates
for RI-AR2 vs Linear Growth AR2

Y2ˆ ON
Y1ˆ 0.413 0.045 9.210

Y3ˆ ON
Y2ˆ 0.307 0.046 6.683
Y1ˆ 0.137 0.041 3.345

Y4ˆ ON
Y3ˆ 0.232 0.050 4.674
Y2ˆ 0.128 0.046 2.795

Y5ˆ ON
Y4ˆ 0.179 0.050 3.563
Y3ˆ 0.068 0.052 1.296

Y6ˆ ON
Y5ˆ 0.215 0.062 3.484
Y4ˆ 0.105 0.048 2.171

Y7ˆ ON
Y6ˆ 0.326 0.049 6.596
Y5ˆ 0.148 0.055 2.703

Y8ˆ ON
Y7ˆ 0.199 0.046 4.348
Y6ˆ 0.099 0.050 1.984

Variances
I 5.442 0.276 19.705

Y2ˆ ON
Y1ˆ 0.373 0.081 4.618

Y3ˆ ON
Y2ˆ 0.324 0.050 6.472
Y1ˆ 0.156 0.054 2.906

Y4ˆ ON
Y3ˆ 0.261 0.047 5.529
Y2ˆ 0.152 0.050 3.047

Y5ˆ ON
Y4ˆ 0.205 0.042 4.930
Y3ˆ 0.081 0.045 1.812

Y6ˆ ON
Y5ˆ 0.130 0.052 2.481
Y4ˆ 0.088 0.040 2.219

Y7ˆ ON
Y6ˆ 0.153 0.075 2.034
Y5ˆ 0.084 0.045 1.843

Y8ˆ ON
Y7ˆ -0.183 0.151 -1.211
Y6ˆ -0.136 0.105 -1.291

S WITH
I -0.158 0.361 -0.438

Means
I 4.896 0.077 63.820
S -0.505 0.052 -9.626

Variances
I 5.340 0.503 10.610
S 1.427 0.468 3.048

Covariance Contributions in the Linear Growth AR2 Model
for Stress

Cov(ys,yt) = V(i)+V(s)(xsxt)+Cov(i,s)(xs +xt)+Cov(yˆs,yˆt)

Estimated Cov(ys,yt) is obtained from Residual or TECH4

Estimated Cov(yˆs,yˆt) is obtained from TECH4

Using y4 and y5 as an example:

Cov = 5.340+1.427∗0.5∗0.7−0.158∗ (0.5+0.7)+1.038

= 5.340+0.499−0.1896+1.038

Corr = 0.511︸ ︷︷ ︸
i

+0.050−0.019︸ ︷︷ ︸
s

+0.103︸ ︷︷ ︸
AR

This shows that the contribution from the random slope is
relatively small

Analysis of Abstinence: Comparing Categorical Model Fit
for RI-AR2 vs Linear Growth AR2 Using Bayes

Week 1 - Week 16 (N=1375, T=8)

Model # par’s PPP/ # Significant Residuals Comment
Resp Pattern (obs. freq) Bivar

1. Unrestricted 36 0.520 1 (38; Z=1.98)* 0 Good fit
2. AR2 21 0.474 0 0 Good fit
3. RI-AR2 22 0.472 0 0 Good fit
4. Linear growth 18 0.436 0 1 Good fit

AR2

* 3rd most frequent pattern

Model 3: Estimated V(i) smaller than its SE (posterior SD), suggesting
that the trait is not substantial so that Model 2 is sufficient

Model 4: Estimated V(i) = 0.256 (.096), V(s) = 0.190 (.076),
correlation between i and s = -0.409, mean of s = 0.109 (.043) which
means that the probability of non-abstinence increases over time

Mplus Input for Linear Growth AR2 for Abstinence

USEVARIABLES = z1-z8;
CATEGORICAL = z1-z8;

DEFINE:
CUTPOINT z1-z8 (0.5);

ANALYSIS:
ESTIMATOR = BAYES;
BITERATIONS = (2000);
THIN = 100;
PROCESSORS = 8;

MODEL:
i s | z1@0 z2@.1 z3@.3 z4@.5
z5@.7 z6@.9 z7@1.1 z8@1.5;
z2ˆ-z8ˆ PON z1ˆ-z7ˆ;
z3ˆ-z8ˆ PON z1ˆ-z6ˆ;
! Fix residual variances to avoid
! growth model Theta param’n
! default of last T-1 free
z1-z8@1;

OUTPUT:
STANDARDIZED RESIDUAL
TECH8 TECH10;

PLOT:
TYPE = PLOT3;

Bivariate Analysis of Stress and Abstinence:
Comparing Model Fit for RI-AR2 vs Linear Growth AR2

for Both Processes, Week 1 - Week 16 (N=1375, T=8)
Bayes and WLSMV Estimation

Model # par’s PPP/ # Significant Residuals Comment
χ2 Resp Pattern (obs. freq) Bivar

1. RI-AR2 75 0.299 0 0 Good fit
RI-AR2* 75 χ2(69)=83 0 0 Good fit

(.1218)
2. Linear Growth 66 0.144 0 1 Good fit

AR2
Linear Growth 66 χ2(78)=107 0 0 Good fit
AR2* (.0151) but npd

* denotes WLSMV

To achieve convergence with Bayes, the linear growth model fixes the
variance of the slope of abstinence at 0 (actually 0.1) and
corresponding covariances. The WLSMV run gives a npd solution

Bivariate Analysis of Stress and Abstinence: Comparing
Cross-Lagged Estimates for RI-AR2 vs Linear Growth AR2

Agreement in terms of # significant cross-lagged effects for the
two models:

Y ON Z: 7 out of 7
Z ON Y: 0/1 out of 7

Non-abstinence has a significant effect on increase of Stress, not
the other way around

Including a trend using linear growth is not essential in these
data from the point of view of cross-lagged effects

Allowing for a Trend Using Longitudinal LCA:
Growth Modeling with Flexible Curve Shapes

c

y1 y2 y3 y4 y5 y6 y7 y8

LCA typically used for cross-sectional analysis but here used for
longitudinal analysis
The latent class variable serves as a non-parametric
representation of a combination of growth factors: i, s, q
ML and Bayes can be used

With Bayes, label switching is a threat (CHAIN=1 helpful)
ML can add residual covariances (RESCOV parameterization)

Bayes can add AR but convergence problems due to mixture are
possible

Residual Associations

c

y1 y2 y3 y4 y5 y6 y7 y8

Parameterization = Rescov
Asparouhov & Muthén (2015). Residual associations in latent
class and latent transition analysis. Structural Equation Modeling:
A Multidisciplinary Journal, 22:2, 169-177

CLASSES = c(2);
ANALYSIS:

TYPE = MIXTURE;
ESTIMATOR = MLR;
PROCESSORS = 8;
STARTS = 80 20;
PARAMETERIZATION = RESCOV;

MODEL:
%OVERALL%
y2-y8 PWITH y1-y7;

OUTPUT:
TECH10;

PLOT:
TYPE = PLOT3;
SERIES = y1-y8(*);

LCA MLR Model Fit: Abstinence

# Sig Standardized Residuals

Model # par’s LL BIC Response Patterns Bivariate Comments

0. Linear Growth 18 0 0 Good fit
AR2, Bayes

1. LCA 2c 17 -5203 10528 18 (all 0’s) 32 Poor fit

2. LCA 3c 26 -4956 10100 7 (all 0’s) 24 Poor fit

3. LCA 4c 35 -4794 9841 4 (19) 2 OK fit

4. LCA 5c 44 -4774 9865 3 (13) 0 0 Good fit

5. LCA 2 rescov 24 -4852 9878 5 (all 0’s) 3 Poor fit

6. LCA 3c rescov 33 -4787 9813 2 (22) 1 Good fit

7. LCA 4c rescov 42 -4763 9830 0 0 Good fit

LCA 3c with Rescov: MLR Estimated Curves
for Non-Abstinence Probability (Entropy = 0.75)

Classes: low (30%), medium (29%), high (41%)

4-class solution splits the medium class into two

Interpreting the Latent Classes

Classes: low (30%), medium (29%), high (41%)

The 3 classes can be seen as a non-parametric representation of a
random intercept instead of the normal distribution specification of the
RI-AR model and the linear growth model

Are the 3 classes substantively meaningful? - How do they relate to
antecedents?

9 treatment groups of size N=150. First group placebo

LCA 3c with Rescov: Mplus Input for Trajectory Classes
Related to 9 Treatment Groups Using 8 Dummy Variables

C ON X carried out using AUXILIARY R3STEP

USEVARIABLES = z1-z8 x2-x9;
CATEGORICAL = z1-z8;
CLASSES = c(3);
AUXILIARY = x2-x9(R3STEP);

DEFINE:
CUTPOINT z1-z8 (0.5);
! ccell = 1 is the placebo group
IF(ccell EQUAL 2)THEN x2=1 ELSE x2=0;
IF(ccell EQUAL 3)THEN x3=1 ELSE x3=0;
IF(ccell EQUAL 4)THEN x4=1 ELSE x4=0;
IF(ccell EQUAL 5)THEN x5=1 ELSE x5=0;
IF(ccell EQUAL 6)THEN x6=1 ELSE x6=0;
IF(ccell EQUAL 7)THEN x7=1 ELSE x7=0;
IF(ccell EQUAL 8)THEN x8=1 ELSE x8=0;
IF(ccell EQUAL 9)THEN x9=1 ELSE x9=0;

ANALYSIS:
TYPE = MIXTURE;
ESTIMATOR = MLR;
PROCESSORS = 8;
STARTS = 80 20;
PARAMETERIZATION = RESCOV;

MODEL:
%OVERALL%
z2-z8 PWITH z1-z7;

OUTPUT:
TECH10;

PLOT:
TYPE = PLOT3;
SERIES = z1-z8(*);

Bengt Muthén RI-CLPM 149/ 162



LCA 3c with Rescov: Trajectory Classes
Related to Treatment Groups

C ON X results using the low class as a comparison class:

Two significant effects: The high class regressed on X3 and X4
has negative effects, i.e., lowering the odds of being in the high
class relative to being in the low class (can be translated into
probabilities as shown in UG chapter 14)
X3 = naltrexone, X4 = naltrexone + acamprosate

LCA-Rescov research question: How to connect to RI-AR modeling
and the bivariate RI-CLPM?

Growth Mixture Modeling (GMM)

i

y1 y2 y3 y4 y5 y6 y7 y8

c

s

Asparouhov-Muthén (2009): http://www.statmodel.com/
download/ChapmanHall06V24.pdf

Particular growth curve shapes chosen beforehand or via LCA

Continuous latent growth factors: i, s, q
Growth factor means drive the probability trajectories while
thresholds are time- and class-invariant
Special case with zero growth factor variances: Latent class
growth analysis (LCGA)

Residual correlations not yet accessible for categorical outcomes:
ML with Rescov not available with random effects
Bayes with AR difficult for mixtures due to no ML starts
(STVALUES=ML) and potential label switching

Section 14 Distal Outcomes

Assessing Treatment Effects: Proximal and Distal Outcomes
in Bivariate Cross-Lagged Panel Modeling for
Stress and Abstinence in the COMBINE Data

Proximal outcome (primary endpoint): Week 16 abstinence
Multiple-group Bayesian analysis using Knownclass for placebo
and treatment groups
Bivariate Linear Growth AR2 model for stress and abstinence
week 1 - week 16 (T=8) from slide 142
Centering the time scores at week 16 to assess primary outcome
effect via the means of the random intercept
Fixing the random intercept means of the placebo group at zero
as a comparison group

Distal (long-term secondary) outcome: Week 52 abstinence
Regressing the week 52 abstinence outcome on the random
intercepts with group-invariant slope to find the treatment effect
by the comparison of the week 52 abstinence probability for the
treatment groups and the placebo group

Mplus Input for
Assessing ProximalTreatment Effects at Week 16

Bivariate Linear Growth AR2 Model
Treatment Effect Assessed by IY and IZ Means

8 treatment groups and placebo group are represented as 9 classes
using Knownclass in a Type=Mixture analysis

USEVARIABLES = y1-y8 z1-z8;
CATEGORICAL = z1-z8;
CLASSES = c(9);
KNOWNCLASS = c(cCell = 1-9);

DEFINE:
CUTPOINT z1-z8 (0.5);

ANALYSIS:
TYPE = MIXTURE;
ESTIMATOR = BAYES;
BITERATIONS = (5000);
THIN = 10;
PROCESSORS = 8;

Mplus Input Continued:
Assessing ProximalTreatment Effects at Week 16
Treatment Effect Assessed by IY and IZ Means

MODEL:
%OVERALL%
iy sy | y1@-1.5 y2@-1.4 y3@-1.2 y4@-1 y5@-
.8 y6@-.6 y7@-.4 y8@0;
iz sz | z1@-1.5 z2@-1.4 z3@-1.2 z4@-1 z5@-.8
z6@-.6 z7@-.4 z8@0;
y2ˆ-y8ˆ PON y1ˆ-y7ˆ;
y3ˆ-y8ˆ PON y1ˆ-y6ˆ;
z2ˆ-z8ˆ PON z1ˆ-z7ˆ;
z3ˆ-z8ˆ PON z1ˆ-z6ˆ;
z1-z8@1;
y2ˆ-y8ˆ PON z1ˆ-z7ˆ;
z2ˆ-z8ˆ PON y1ˆ-y7ˆ;
y1ˆ-y8ˆ PWITH z1ˆ-z8ˆ;

sz@0.1;
sz WITH iz@0;
sz WITH iy@0;
sz WITH sy@0;
! changing the default:
[y1-y8] (int);
%c#1% ! Placebo group
[iy@0 iz@0];
%c#9%
! Avoiding mixture default
! of zero means in last class:
[iy-sz];

OUTPUT:
STANDARDIZED RESIDUAL
TECH1 TECH8 TECH10;

PLOT:
TYPE = PLOT3;

Assessing Proximal Treatment Effects: Results

The means for the IZ abstinence random intercept show significant
treatment effects on week 16 abstinence for 4 groups:

3: naltrexone
4: naltrexone + acamprosate
5: placebo + behavioral intervention
8: naltrexone + acamprosate + behavioral intervention

No significant treatment effect on the IY stress random intercept means

Mplus Input for Assessing Treatment Effects on Week 52

! z9 is week 52 abstinence
USEVARIABLES = y1-y8 z1-z9;
CATEGORICAL = z1-z9;
CLASSES = c(9);
KNOWNCLASS = c(cCell = 1-9);

DEFINE:
CUTPOINT z1-z9 (0.5);

ANALYSIS:
ESTIMATOR = BAYES;
BITERATIONS = (2000);
THIN = 100;
PROCESSORS = 8;
TYPE = MIXTURE;

MODEL:
%OVERALL%
iy sy | y1@-1.5 y2@-1.4 y3@-1.2
y4@-1 y5@-.8 y6@-.6 y7@-.4 y8@0;
iz sz | z1@-1.5 z2@-1.4 z3@-1.2 z4@-
1 z5@-.8 z6@-.6 z7@-.4 z8@0;
y2ˆ-y8ˆ PON y1ˆ-y7ˆ;
y3ˆ-y8ˆ PON y1ˆ-y6ˆ;
z2ˆ-z8ˆ PON z1ˆ-z7ˆ;
z3ˆ-z8ˆ PON z1ˆ-z6ˆ;
z1-z8@1;
y2ˆ-y8ˆ PON z1ˆ-z7ˆ;
z2ˆ-z8ˆ PON y1ˆ-y7ˆ;
y1ˆ-y8ˆ PWITH z1ˆ-z8ˆ;
sz@0.1;
sz WITH iz@0;
sz WITH iy@0;
sz WITH sy@0;
[y1-y8] (int);
z9 ON iz (b);
[iy@0];
[sy] (slopey);
[sz] (slopez);

Mplus Input Continued

%c#1%
[iy@0 iz@0];
[z9$1] (t1);
%c#2%
[z9$1] (t2);
[iz] (m2);
%c#3%
[z9$1] (t3);
[iz] (m3);
%c#4%
[z9$1] (t4);
[iz] (m4);
%c#5%
[z9$1] (t5);
[iz] (m5);
%c#6%
[z9$1] (t6);
[iz] (m6);
%c#7%
[z9$1] (t7);
[iz] (m7);

%c#8%
[z9$1] (t8);
[iz] (m8);
%c#9%
[z9$1] (t9);
[iz] (m9);

MODEL
CONSTRAINT:

NEW(d2-d9);
! probit for Z9,
! = -t + b*m:
! probit for placebo
! = -t1 (m1=0)
! probit diffs for tx - placebo:
d2 = -t2+b*m2+t1;
d3 = -t3+b*m3+t1;
d4 = -t4+b*m4+t1;
d5 = -t5+b*m5+t1;
d6 = -t6+b*m6+t1;
d7 = -t7+b*m7+t1;
d8 = -t8+b*m8+t1;
d9 = -t9+b*m9+t1;

Bengt Muthén RI-CLPM 158/ 162



Assessing Distal Treatment Effects: Results

No significant treatment effects at follow-up week 52:

The Z9* means (d2 - d9 in Model Constraint) for the treatment
groups are not significantly different from that of the placebo
group

The random intercept for abstinence is a significant predictor of Z9

A higher random intercept corresponds to a higher probability of
non-abstinence

Extensions:

Two-part ordinal both for growth part and distal outcome
Using pre-treatment data and a pre-treatment baseline random
intercept, it is possible to investigate treatment-baseline
interaction

Other Non-Continuous Outcomes

Other non-continuous outcome types can be accomodated such as
semi-continuous variables

Data set 3 negative affect score summing over a set of items:

Can be analyzed by Bayes using censored or two-part RI-AR models
and the new hat approach. Right tail of the distribution treated as
continuous

